Rules for integrands of the form $\left(a+b\,\mathrm{Sin}[e+f\,x]^2\right)^p\left(A+B\,\mathrm{Sin}[e+f\,x]^2\right)$

1. $\int\left(a+b\,\mathrm{Sin}[e+f\,x]^2\right)^p\left(A+B\,\mathrm{Sin}[e+f\,x]^2\right)dx$ when $p>0$

   1: $\int\left(a+b\,\mathrm{Sin}[e+f\,x]^2\right)\left(A+B\,\mathrm{Sin}[e+f\,x]^2\right)dx$

Derivation: Algebraic expansion

Basis: $(a+b\,z)\,(A+B\,z)==\frac{1}{8}\,(4\,A\,(2\,a+b)+B\,(4\,a+3\,b))-\frac{1}{8}\,(4\,A\,b+B\,(4\,a+3\,b))\,(1-2\,z)-\frac{1}{4}\,b\,B\,z\,(3-4\,z)$

Rule:

$$\int\left(a+b\,\mathrm{Sin}[e+f\,x]^2\right)\left(A+B\,\mathrm{Sin}[e+f\,x]^2\right)dx\to$$
$$\frac{(4\,A\,(2\,a+b)+B\,(4\,a+3\,b))\,x}{8}-\frac{(4\,A\,b+B\,(4\,a+3\,b))\,\mathrm{Cos}[e+f\,x]\,\mathrm{Sin}[e+f\,x]}{8\,f}-\frac{b\,B\,\mathrm{Cos}[e+f\,x]\,\mathrm{Sin}[e+f\,x]^3}{4\,f}$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_]^2)*(A_.+B_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  (4*A*(2*a+b)+B*(4*a+3*b))*x/8 -
  (4*A*b+B*(4*a+3*b))*Cos[e+f*x]*Sin[e+f*x]/(8*f) -
  b*B*Cos[e+f*x]*Sin[e+f*x]^3/(4*f) /;
FreeQ[{a,b,e,f,A,B},x]
```

2: $\int \left(a+b\,\text{Sin}[e+f\,x]^2\right)^p \left(A+B\,\text{Sin}[e+f\,x]^2\right) dx$ when $p>0$

Rule: If $p>0$, then

$$\int \left(a+b\,\text{Sin}[e+f\,x]^2\right)^p \left(A+B\,\text{Sin}[e+f\,x]^2\right) dx \longrightarrow$$

$$-\frac{B\,\text{Cos}[e+f\,x]\,\text{Sin}[e+f\,x]\left(a+b\,\text{Sin}[e+f\,x]^2\right)^p}{2\,f\,(p+1)}+$$

$$\frac{1}{2\,(p+1)}\int \left(a+b\,\text{Sin}[e+f\,x]^2\right)^{p-1}\left(a\,B+2\,a\,A\,(p+1)+(2\,A\,b\,(p+1)+B\,(b+2\,a\,p+2\,b\,p))\,\text{Sin}[e+f\,x]^2\right) dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_]^2)^p_*(A_.+B_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  -B*Cos[e+f*x]*Sin[e+f*x]*(a+b*Sin[e+f*x]^2)^p/(2*f*(p+1)) +
  1/(2*(p+1))*Int[(a+b*Sin[e+f*x]^2)^(p-1)*
    Simp[a*B+2*a*A*(p+1)+(2*A*b*(p+1)+B*(b+2*a*p+2*b*p))*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,e,f,A,B},x] && GtQ[p,0]
```

2. $\int (a + b\,\mathrm{Sin}[e + f\,x]^2)^p\,(A + B\,\mathrm{Sin}[e + f\,x]^2)\,dx$ when $p < 0$

1: $\int \frac{A + B\,\mathrm{Sin}[c + d\,x]^2}{a + b\,\mathrm{Sin}[e + f\,x]^2}\,dx$

Derivation: Algebraic expansion

Basis: $\frac{A+B\,z}{a+b\,z} == \frac{B}{b} + \frac{A\,b-a\,B}{b\,(a+b\,z)}$

Rule:

$$\int \frac{A + B\,\mathrm{Sin}[c + d\,x]^2}{a + b\,\mathrm{Sin}[e + f\,x]^2}\,dx \longrightarrow \frac{B\,x}{b} + \frac{A\,b - a\,B}{b} \int \frac{1}{a + b\,\mathrm{Sin}[e + f\,x]^2}\,dx$$

Program code:

```
Int[(A_.+B_.*sin[e_.+f_.*x_]^2)/(a_+b_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  B*x/b + (A*b-a*B)/b*Int[1/(a+b*Sin[e+f*x]^2),x] /;
FreeQ[{a,b,e,f,A,B},x]
```

2: $\int \frac{A + B \operatorname{Sin}[c + d x]^2}{\sqrt{a + b \operatorname{Sin}[e + f x]^2}} \, dx$

Derivation: Algebraic expansion

Basis: $\frac{A+B \operatorname{Sin}[z]^2}{\sqrt{a+b \operatorname{Sin}[z]^2}} = \frac{B \sqrt{a+b \operatorname{Sin}[z]^2}}{b} + \frac{A b-a B}{b \sqrt{a+b \operatorname{Sin}[z]^2}}$

Rule:

$$\int \frac{A + B \operatorname{Sin}[c + d x]^2}{\sqrt{a + b \operatorname{Sin}[e + f x]^2}} \, dx \longrightarrow \frac{B}{b} \int \sqrt{a + b \operatorname{Sin}[e + f x]^2} \, dx + \frac{A b - a B}{b} \int \frac{1}{\sqrt{a + b \operatorname{Sin}[e + f x]^2}} \, dx$$

Program code:

```
Int[(A_.+B_.*sin[e_.+f_.*x_]^2)/Sqrt[a_+b_.*sin[e_.+f_.*x_]^2],x_Symbol] :=
  B/b*Int[Sqrt[a+b*Sin[e+f*x]^2],x] + (A*b-a*B)/b*Int[1/Sqrt[a+b*Sin[e+f*x]^2],x] /;
FreeQ[{a,b,e,f,A,B},x]
```

3: $\int \left(a + b \operatorname{Sin}[e + f x]^2\right)^p \left(A + B \operatorname{Sin}[e + f x]^2\right) dx$ when $p < -1 \land a + b \neq 0$

Rule: If $p < -1 \land a + b \neq 0$, then

$$\int \left(a + b \operatorname{Sin}[e + f x]^2\right)^p \left(A + B \operatorname{Sin}[e + f x]^2\right) dx \longrightarrow$$
$$-\frac{(A b - a B) \operatorname{Cos}[e + f x] \operatorname{Sin}[e + f x] \left(a + b \operatorname{Sin}[e + f x]^2\right)^{p+1}}{2 a f (a + b) (p + 1)} - \frac{1}{2 a (a + b) (p + 1)} \int \left(a + b \operatorname{Sin}[e + f x]^2\right)^{p+1} \left(a B - A (2 a (p + 1) + b (2 p + 3)) + 2 (A b - a B) (p + 2) \operatorname{Sin}[e + f x]^2\right) dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_]^2)^p_*(A_.+B_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  -(A*b-a*B)*Cos[e+f*x]*Sin[e+f*x]*(a+b*Sin[e+f*x]^2)^(p+1)/(2*a*f*(a+b)*(p+1)) -
  1/(2*a*(a+b)*(p+1))*Int[(a+b*Sin[e+f*x]^2)^(p+1)*
    Simp[a*B-A*(2*a*(p+1)+b*(2*p+3))+2*(A*b-a*B)*(p+2)*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,e,f,A,B},x] && LtQ[p,-1] && NeQ[a+b,0]
```

3: $\int \left(a + b \operatorname{Sin}[e + f x]^2\right)^p \left(A + B \operatorname{Sin}[e + f x]^2\right) dx$ when $p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $A + B \operatorname{Sin}[z]^2 == \frac{A + (A + B) \operatorname{Tan}[z]^2}{1 + \operatorname{Tan}[z]^2}$

Basis: $\partial_x \frac{\left(a + b \operatorname{Sin}[e + f x]^2\right)^p \left(\operatorname{Sec}[e + f x]^2\right)^p}{\left(a + (a + b) \operatorname{Tan}[e + f x]^2\right)^p} == 0$

Basis: $F[\operatorname{Tan}[e + f x]] == \frac{1}{f} \operatorname{Subst}\left[\frac{F[x]}{1 + x^2}, x, \operatorname{Tan}[e + f x]\right] \partial_x \operatorname{Tan}[e + f x]$

Rule: If $p \notin \mathbb{Z}$, then

$$\int \left(a+b\,\text{Sin}[e+f\,x]^2\right)^p \left(A+B\,\text{Sin}[e+f\,x]^2\right) dx \longrightarrow \frac{\left(a+b\,\text{Sin}[e+f\,x]^2\right)^p \left(\text{Sec}[e+f\,x]^2\right)^p}{\left(a+(a+b)\,\text{Tan}[e+f\,x]^2\right)^p} \int \frac{\left(a+(a+b)\,\text{Tan}[e+f\,x]^2\right)^p \left(A+(A+B)\,\text{Tan}[e+f\,x]^2\right)}{\left(1+\text{Tan}[e+f\,x]^2\right)^{p+1}} dx$$

$$\longrightarrow \frac{\left(a+b\,\text{Sin}[e+f\,x]^2\right)^p \left(\text{Sec}[e+f\,x]^2\right)^p}{f\left(a+(a+b)\,\text{Tan}[e+f\,x]^2\right)^p} \,\text{Subst}\left[\int \frac{\left(a+(a+b)\,x^2\right)^p \left(A+(A+B)\,x^2\right)}{\left(1+x^2\right)^{p+2}} dx,\, x,\, \text{Tan}[e+f\,x]\right]$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_]^2)^p_*(A_.+B_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff*(a+b*Sin[e+f*x]^2)^p*(Sec[e+f*x]^2)^p/(f*(a+(a+b)*Tan[e+f*x]^2)^p)*
    Subst[Int[(a+(a+b)*ff^2*x^2)^p*(A+(A+B)*ff^2*x^2)/(1+ff^2*x^2)^(p+2),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,e,f,A,B},x] && Not[IntegerQ[p]]
```

Rules for integrands of the form $u\left(a+b\,\text{Sin}[e+f\,x]^2\right)^p$

1. $\int u\left(a+b\,\text{Sin}[e+f\,x]^2\right)^p\,dx$ when $a+b==0$

1: $\int u\left(a+b\,\text{Sin}[e+f\,x]^2\right)^p\,dx$ when $a+b==0\,\land\,p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $a+b==0$, then $a+b\,\text{Sin}[z]^2==a\,\text{Cos}[z]^2$

Rule: If $a+b==0\,\land\,p\in\mathbb{Z}$, then

$$\int u\left(a+b\,\text{Sin}[e+f\,x]^2\right)^p\,dx\;\longrightarrow\;a^p\int u\,\text{Cos}[e+f\,x]^{2p}\,dx$$

Program code:

```
Int[u_.*(a_+b_.*sin[e_.+f_.*x_]^2)^p_,x_Symbol] :=
  a^p*Int[ActivateTrig[u*cos[e+f*x]^(2*p)],x] /;
FreeQ[{a,b,e,f,p},x] && EqQ[a+b,0] && IntegerQ[p]
```

2: $\int u\,(a+b\,\mathrm{Sin}[e+f\,x]^2)^p\,dx$ when $a+b==0$

Derivation: Algebraic simplification

Basis: If $a+b==0$, then $a+b\,\mathrm{Sin}[z]^2==a\,\mathrm{Cos}[z]^2$

Rule: If $a+b==0$, then

$$\int u\,(a+b\,\mathrm{Sin}[e+f\,x]^2)^p\,dx \longrightarrow \int u\,(a\,\mathrm{Cos}[e+f\,x]^2)^p\,dx$$

Program code:

```
Int[u_.*(a_+b_.*sin[e_.+f_.*x_]^2)^p_,x_Symbol] :=
  Int[ActivateTrig[u*(a*cos[e+f*x]^2)^p],x] /;
FreeQ[{a,b,e,f,p},x] && EqQ[a+b,0]
```

2. $\int \left(a+b\,\text{Sin}[e+f\,x]^2\right)^p\,dx$

1. $\int \left(a+b\,\text{Sin}[e+f\,x]^2\right)^p\,dx$ when $a+b\neq 0\land p>0$

1. $\int \sqrt{a+b\,\text{Sin}[e+f\,x]^2}\,dx$

1: $\int \sqrt{a+b\,\text{Sin}[e+f\,x]^2}\,dx$ when $a>0$

Rule: If $a>0$, then

$$\int \sqrt{a+b\,\text{Sin}[e+f\,x]^2}\,dx \to \frac{\sqrt{a}}{f}\,\text{EllipticE}\left[e+f\,x,-\frac{b}{a}\right]$$

Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]^2],x_Symbol] :=
  Sqrt[a]/f*EllipticE[e+f*x,-b/a] /;
FreeQ[{a,b,e,f},x] && GtQ[a,0]
```

2: $\int \sqrt{a+b\,\text{Sin}[e+f\,x]^2}\,dx$ when $a\not>0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]^2}}{\sqrt{1+\frac{b\,\text{Sin}[e+f\,x]^2}{a}}} == 0$

Rule: If $a\not>0$, then

$$\int \sqrt{a+b\,\text{Sin}[e+f\,x]^2}\,dx \longrightarrow \frac{\sqrt{a+b\,\text{Sin}[e+f\,x]^2}}{\sqrt{1+\frac{b\,\text{Sin}[e+f\,x]^2}{a}}}\int \sqrt{1+\frac{b\,\text{Sin}[e+f\,x]^2}{a}}\,dx$$

- ## Program code:

```
Int[Sqrt[a_+b_.*sin[e_.+f_.*x_]^2],x_Symbol] :=
  Sqrt[a+b*Sin[e+f*x]^2]/Sqrt[1+b*Sin[e+f*x]^2/a]*Int[Sqrt[1+(b*Sin[e+f*x]^2)/a],x] /;
FreeQ[{a,b,e,f},x] && Not[GtQ[a,0]]
```

2: $\int \left(a+b\,\text{Sin}[e+f\,x]^2\right)^2 dx$

- ## Derivation: Algebraic expansion

- ## Basis: $(a+b\,z)^2 == \frac{1}{8}\left(8\,a^2+8\,a\,b+3\,b^2\right)-\frac{b}{8}\,(8\,a+3\,b)\,(1-2\,z)-\frac{1}{4}\,b^2\,(3-4\,z)\,z$

Rule:

$$\int \left(a+b\,\text{Sin}[e+f\,x]^2\right)^2 dx \longrightarrow$$
$$\frac{\left(8\,a^2+8\,a\,b+3\,b^2\right)x}{8}-\frac{b\,(8\,a+3\,b)\,\text{Cos}[e+f\,x]\,\text{Sin}[e+f\,x]}{8\,f}-\frac{b^2\,\text{Cos}[e+f\,x]\,\text{Sin}[e+f\,x]^3}{4\,f}$$

- ## Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_]^2)^2,x_Symbol] :=
  (8*a^2+8*a*b+3*b^2)*x/8 -
  b*(8*a+3*b)*Cos[e+f*x]*Sin[e+f*x]/(8*f) -
  b^2*Cos[e+f*x]*Sin[e+f*x]^3/(4*f) /;
FreeQ[{a,b,e,f},x]
```

3: $\int \left(a + b \operatorname{Sin}[e + f x]^2\right)^p dx$ when $a + b \neq 0 \land p > 1$

Rule: If $a + b \neq 0 \land p > 1$, then

$$\int \left(a + b \operatorname{Sin}[e + f x]^2\right)^p dx \longrightarrow$$
$$-\frac{b \operatorname{Cos}[e + f x] \operatorname{Sin}[e + f x] \left(a + b \operatorname{Sin}[e + f x]^2\right)^{p-1}}{2 f p} +$$
$$\frac{1}{2 p} \int \left(a + b \operatorname{Sin}[e + f x]^2\right)^{p-2} \left(a (b + 2 a p) + b (2 a + b) (2 p - 1) \operatorname{Sin}[e + f x]^2\right) dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_]^2)^p_,x_Symbol] :=
  -b*Cos[e+f*x]*Sin[e+f*x]*(a+b*Sin[e+f*x]^2)^(p-1)/(2*f*p) +
  1/(2*p)*Int[(a+b*Sin[e+f*x]^2)^(p-2)*Simp[a*(b+2*a*p)+b*(2*a+b)*(2*p-1)*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,e,f},x] && NeQ[a+b,0] && GtQ[p,1]
```

2. $\int \left(a + b \operatorname{Sin}[e + f x]^2\right)^p dx$ when $a + b \neq 0 \land p < 0$

1: $\int \frac{1}{a + b \operatorname{Sin}[e + f x]^2} dx$

Derivation: Integration by substitution

Basis: $\operatorname{Sin}[z]^2 = \frac{\operatorname{Tan}[z]^2}{1 + \operatorname{Tan}[z]^2}$

Basis: $F\left[\operatorname{Sin}[e + f x]^2\right] = \frac{1}{f} \operatorname{Subst}\left[\frac{F\left[\frac{x^2}{1+x^2}\right]}{1+x^2}, x, \operatorname{Tan}[e + f x]\right] \partial_x \operatorname{Tan}[e + f x]$

Rule: If $p \in \mathbb{Z}$, then

$$\int \frac{1}{a+b\,\mathrm{Sin}[e+f\,x]^2}\,dx \longrightarrow \frac{1}{f}\,\mathrm{Subst}\Big[\int \frac{1}{a+(a+b)\,x^2}\,dx,\ x,\ \mathrm{Tan}[e+f\,x]\Big]$$

Program code:

```
Int[1/(a_+b_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff/f*Subst[Int[1/(a+(a+b)*ff^2*x^2),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,e,f},x]
```

2. $\int \frac{1}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]^2}}\,dx$

1: $\int \frac{1}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]^2}}\,dx$ when $a>0$

Rule: If $a>0$, then

$$\int \frac{1}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]^2}}\,dx \longrightarrow \frac{1}{\sqrt{a}\,f}\,\mathrm{EllipticF}\Big[e+f\,x,\ -\frac{b}{a}\Big]$$

Program code:

```
Int[1/Sqrt[a_+b_.*sin[e_.+f_.*x_]^2],x_Symbol] :=
  1/(Sqrt[a]*f)*EllipticF[e+f*x,-b/a] /;
FreeQ[{a,b,e,f},x] && GtQ[a,0]
```

2: $\int \frac{1}{\sqrt{a+b\,\mathrm{Sin}[e+f\,x]^2}}\,dx$ when $a \ngtr 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{1+\frac{b \operatorname{Sin}[e+f x]^2}{a}}}{\sqrt{a+b \operatorname{Sin}[e+f x]^2}} == 0$

Rule: If $a \ngtr 0$, then

$$\int \frac{1}{\sqrt{a+b \operatorname{Sin}[e+f x]^2}} \, dx \longrightarrow \frac{\sqrt{1+\frac{b \operatorname{Sin}[e+f x]^2}{a}}}{\sqrt{a+b \operatorname{Sin}[e+f x]^2}} \int \frac{1}{\sqrt{1+\frac{b \operatorname{Sin}[e+f x]^2}{a}}} \, dx$$

Program code:

```
Int[1/Sqrt[a_+b_.*sin[e_.+f_.*x_]^2],x_Symbol] :=
  Sqrt[1+b*Sin[e+f*x]^2/a]/Sqrt[a+b*Sin[e+f*x]^2]*Int[1/Sqrt[1+(b*Sin[e+f*x]^2)/a],x] /;
FreeQ[{a,b,e,f},x] && Not[GtQ[a,0]]
```

3: $\int \left(a+b \operatorname{Sin}[e+f x]^2\right)^p \, dx$ when $a+b \neq 0 \land p < -1$

Rule: If $a+b \neq 0 \land p < -1$, then

$$\int \left(a+b \operatorname{Sin}[e+f x]^2\right)^p \, dx \longrightarrow$$
$$-\frac{b \operatorname{Cos}[e+f x] \operatorname{Sin}[e+f x] \left(a+b \operatorname{Sin}[e+f x]^2\right)^{p+1}}{2 a f (p+1) (a+b)} +$$
$$\frac{1}{2 a (p+1) (a+b)} \int \left(a+b \operatorname{Sin}[e+f x]^2\right)^{p+1} \left(2 a (p+1)+b (2 p+3)-2 b (p+2) \operatorname{Sin}[e+f x]^2\right) \, dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_]^2)^p_,x_Symbol] :=
  -b*Cos[e+f*x]*Sin[e+f*x]*(a+b*Sin[e+f*x]^2)^(p+1)/(2*a*f*(p+1)*(a+b)) +
  1/(2*a*(p+1)*(a+b))*Int[(a+b*Sin[e+f*x]^2)^(p+1)*Simp[2*a*(p+1)+b*(2*p+3)-2*b*(p+2)*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,e,f},x] && NeQ[a+b,0] && LtQ[p,-1]
```

3: $\int \left(a + b \sin[e + f x]^2\right)^p \, dx$ when $p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{\sqrt{\cos[e+f x]^2}}{\cos[e+f x]} == 0$

Basis: $\cos[e + f x] \, F[\sin[e + f x]] == \frac{1}{f} \text{Subst}[F[x], x, \sin[e + f x]] \, \partial_x \sin[e + f x]$

Rule: If $p \notin \mathbb{Z}$, then

$$\int \left(a + b \sin[e + f x]^2\right)^p dx \to \frac{\sqrt{\cos[e + f x]^2}}{\cos[e + f x]} \int \frac{\cos[e + f x] \left(a + b \sin[e + f x]^2\right)^p}{\sqrt{1 - \sin[e + f x]^2}} dx$$

$$\to \frac{\sqrt{\cos[e + f x]^2}}{f \cos[e + f x]} \text{Subst}\left[\int \frac{\left(a + b x^2\right)^p}{\sqrt{1 - x^2}} dx, x, \sin[e + f x]\right]$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_]^2)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Sin[e+f*x],x]},
  ff*Sqrt[Cos[e+f*x]^2]/(f*Cos[e+f*x])*Subst[Int[(a+b*ff^2*x^2)^p/Sqrt[1-ff^2*x^2],x],x,Sin[e+f*x]/ff]] /;
FreeQ[{a,b,e,f,p},x] && Not[IntegerQ[p]]
```

3. $\int (d\,\mathrm{Sin}[e+f\,x])^m\,(a+b\,\mathrm{Sin}[e+f\,x]^2)^p\,dx$

1. $\int \mathrm{Sin}[e+f\,x]^m\,(a+b\,\mathrm{Sin}[e+f\,x]^2)^p\,dx$ when $m\in\mathbb{Z}$

1: $\int \mathrm{Sin}[e+f\,x]^m\,(a+b\,\mathrm{Sin}[e+f\,x]^2)^p\,dx$ when $\frac{m-1}{2}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $\mathrm{Sin}[z]^2 == 1-\mathrm{Cos}[z]^2$

Basis: If $\frac{m-1}{2}\in\mathbb{Z}$, then

$\mathrm{Sin}[e+f\,x]^m\,F\left[\mathrm{Sin}[e+f\,x]^2\right] == -\frac{1}{f}\,\mathrm{Subst}\left[\left(1-x^2\right)^{\frac{m-1}{2}}\,F\left[1-x^2\right],\,x,\,\mathrm{Cos}[e+f\,x]\right]\,\partial_x \mathrm{Cos}[e+f\,x]$

Rule: If $\frac{m-1}{2}\in\mathbb{Z}$, then

$$\int \mathrm{Sin}[e+f\,x]^m\,(a+b\,\mathrm{Sin}[e+f\,x]^n)^p\,dx \longrightarrow -\frac{1}{f}\,\mathrm{Subst}\left[\int \left(1-x^2\right)^{\frac{m-1}{2}}\,\left(a+b-b\,x^2\right)^p\,dx,\,x,\,\mathrm{Cos}[e+f\,x]\right]$$

Program code:

```
Int[sin[e_.+f_.*x_]^m_.*(a_+b_.*sin[e_.+f_.*x_]^2)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Cos[e+f*x],x]},
  -ff/f*Subst[Int[(1-ff^2*x^2)^((m-1)/2)*(a+b-b*ff^2*x^2)^p,x],x,Cos[e+f*x]/ff]] /;
FreeQ[{a,b,e,f,p},x] && IntegerQ[(m-1)/2]
```

2. $\int \mathrm{Sin}[e+f\,x]^m \left(a+b\,\mathrm{Sin}[e+f\,x]^2\right)^p dx$ when $\frac{m}{2} \in \mathbb{Z}$

1: $\int \mathrm{Sin}[e+f\,x]^m \left(a+b\,\mathrm{Sin}[e+f\,x]^2\right)^p dx$ when $\frac{m}{2} \in \mathbb{Z} \land p \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: $\mathrm{Sin}[z]^2 == \frac{\mathrm{Tan}[z]^2}{1+\mathrm{Tan}[z]^2}$

Basis: If $\frac{m}{2} \in \mathbb{Z}$, then $\mathrm{Sin}[e+f\,x]^m\,F\left[\mathrm{Sin}[e+f\,x]^2\right] == \frac{1}{f}\,\mathrm{Subst}\left[\frac{x^m\,F\left[\frac{x^2}{1+x^2}\right]}{\left(1+x^2\right)^{m/2+1}},\,x,\,\mathrm{Tan}[e+f\,x]\right]\partial_x \mathrm{Tan}[e+f\,x]$

Rule: If $\frac{m}{2} \in \mathbb{Z} \land p \in \mathbb{Z}$, then

$$\int \mathrm{Sin}[e+f\,x]^m \left(a+b\,\mathrm{Sin}[e+f\,x]^2\right)^p dx \longrightarrow \frac{1}{f}\,\mathrm{Subst}\left[\int \frac{x^m \left(a+(a+b)\,x^2\right)^p}{\left(1+x^2\right)^{m/2+p+1}}\,dx,\,x,\,\mathrm{Tan}[e+f\,x]\right]$$

Program code:

```
Int[sin[e_.+f_.*x_]^m_*(a_+b_.*sin[e_.+f_.*x_]^2)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff^(m+1)/f*Subst[Int[x^m*(a+(a+b)*ff^2*x^2)^p/(1+ff^2*x^2)^(m/2+p+1),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,e,f},x] && IntegerQ[m/2] && IntegerQ[p]
```

2: $\int \mathrm{Sin}[e+f\,x]^m \left(a+b\,\mathrm{Sin}[e+f\,x]^2\right)^p \,dx$ when $\frac{m}{2} \in \mathbb{Z} \land p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{\sqrt{\mathrm{Cos}[e+f\,x]^2}}{\mathrm{Cos}[e+f\,x]} == 0$

Basis: $\mathrm{Cos}[e+f\,x]\,F[\mathrm{Sin}[e+f\,x]] == \frac{1}{f}\,\mathrm{Subst}[F[x],x,\mathrm{Sin}[e+f\,x]]\,\partial_x \mathrm{Sin}[e+f\,x]$

Rule: If $\frac{m}{2} \in \mathbb{Z} \land p \notin \mathbb{Z}$, then

$$\int \mathrm{Sin}[e+f\,x]^m \left(a+b\,\mathrm{Sin}[e+f\,x]^2\right)^p dx \to \frac{\sqrt{\mathrm{Cos}[e+f\,x]^2}}{\mathrm{Cos}[e+f\,x]} \int \frac{\mathrm{Cos}[e+f\,x]\,\mathrm{Sin}[e+f\,x]^m \left(a+b\,\mathrm{Sin}[e+f\,x]^2\right)^p}{\sqrt{1-\mathrm{Sin}[e+f\,x]^2}}\,dx$$

$$\to \frac{\sqrt{\mathrm{Cos}[e+f\,x]^2}}{f\,\mathrm{Cos}[e+f\,x]}\,\mathrm{Subst}\left[\int \frac{x^m \left(a+b\,x^2\right)^p}{\sqrt{1-x^2}}\,dx,\,x,\,\mathrm{Sin}[e+f\,x]\right]$$

Program code:

```
Int[sin[e_.+f_.*x_]^m_*(a_+b_.*sin[e_.+f_.*x_]^2)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Sin[e+f*x],x]},
  ff^(m+1)*Sqrt[Cos[e+f*x]^2]/(f*Cos[e+f*x])*Subst[Int[x^m*(a+b*ff^2*x^2)^p/Sqrt[1-ff^2*x^2],x],x,Sin[e+f*x]/ff]] /;
FreeQ[{a,b,e,f,p},x] && IntegerQ[m/2] && Not[IntegerQ[p]]
```

2: $\int (d\,\text{Sin}[e+f\,x])^m\,(a+b\,\text{Sin}[e+f\,x]^2)^p\,dx$ when $m\notin\mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{(d\,\text{Sin}[e+f\,x])^{m-1}}{(\text{Sin}[e+f\,x]^2)^{\frac{m-1}{2}}} == 0$

Basis: $\text{Sin}[e+f\,x]\,F[\text{Sin}[e+f\,x]^2] == -\frac{1}{f}\,\text{Subst}[F[1-x^2],\,x,\,\text{Cos}[e+f\,x]]\,\partial_x\text{Cos}[e+f\,x]$

Rule: If $m\notin\mathbb{Z}$, then

$$\int (d\,\text{Sin}[e+f\,x])^m\,(a+b\,\text{Sin}[e+f\,x]^2)^p\,dx \rightarrow d\int \text{Sin}[e+f\,x]\,(d\,\text{Sin}[e+f\,x])^{m-1}\,(a+b\,\text{Sin}[e+f\,x]^2)^p\,dx$$

$$\rightarrow \frac{d\,(d\,\text{Sin}[e+f\,x])^{m-1}}{(\text{Sin}[e+f\,x]^2)^{\frac{m-1}{2}}}\int \text{Sin}[e+f\,x]\,(\text{Sin}[e+f\,x]^2)^{\frac{m-1}{2}}\,(a+b\,\text{Sin}[e+f\,x]^2)^p\,dx$$

$$\rightarrow -\frac{d^{2\,\text{IntPart}[\frac{m-1}{2}]+1}\,(d\,\text{Sin}[e+f\,x])^{2\,\text{FracPart}[\frac{m-1}{2}]}}{f\,(\text{Sin}[e+f\,x]^2)^{\text{racPart}[\frac{m-1}{2}]}}\,\text{Subst}\left[\int (1-x^2)^{\frac{m-1}{2}}\,(a+b-b\,x^2)^p\,dx,\,x,\,\text{Cos}[e+f\,x]\right]$$

Program code:

```
Int[(d_.*sin[e_.+f_.*x_])^m_*(a_+b_.*sin[e_.+f_.*x_]^2)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Cos[e+f*x],x]},
  -ff*d^(2*IntPart[(m-1)/2]+1)*(d*Sin[e+f*x])^(2*FracPart[(m-1)/2])/(f*(Sin[e+f*x]^2)^FracPart[(m-1)/2])*
    Subst[Int[(1-ff^2*x^2)^((m-1)/2)*(a+b-b*ff^2*x^2)^p,x],x,Cos[e+f*x]/ff]] /;
FreeQ[{a,b,d,e,f,m,p},x] && Not[IntegerQ[m]]
```

4. $\int (d \operatorname{Cos}[e+f x])^m \left(a+b \operatorname{Sin}[e+f x]^2\right)^p \, dx$

   1. $\int \operatorname{Cos}[e+f x]^m \left(a+b \operatorname{Sin}[e+f x]^2\right)^p \, dx$ when $m \in \mathbb{Z}$

      1: $\int \operatorname{Cos}[e+f x]^m \left(a+b \operatorname{Sin}[e+f x]^2\right)^p \, dx$ when $\frac{m-1}{2} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{m-1}{2} \in \mathbb{Z}$, then

$$\operatorname{Cos}[e+f x]^m F[\operatorname{Sin}[e+f x]] = \frac{1}{f} \operatorname{Subst}\left[\left(1-x^2\right)^{\frac{m-1}{2}} F[x],\ x,\ \operatorname{Sin}[e+f x]\right] \partial_x \operatorname{Sin}[e+f x]$$

Rule: If $\frac{m-1}{2} \in \mathbb{Z}$, then

$$\int \operatorname{Cos}[e+f x]^m \left(a+b \operatorname{Sin}[e+f x]^2\right)^p \, dx \longrightarrow \frac{1}{f} \operatorname{Subst}\left[\int \left(1-x^2\right)^{\frac{m-1}{2}} \left(a+b x^2\right)^p \, dx,\ x,\ \operatorname{Sin}[e+f x]\right]$$

Program code:

```
Int[cos[e_.+f_.*x_]^m_.*(a_+b_.*sin[e_.+f_.*x_]^2)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Sin[e+f*x],x]},
  ff/f*Subst[Int[(1-ff^2*x^2)^((m-1)/2)*(a+b*ff^2*x^2)^p,x],x,Sin[e+f*x]/ff]] /;
FreeQ[{a,b,e,f,p},x] && IntegerQ[(m-1)/2]
```

2. $\int \cos[e+f x]^m \left(a+b \sin[e+f x]^2\right)^p dx$ when $\frac{m}{2} \in \mathbb{Z}$

1: $\int \cos[e+f x]^m \left(a+b \sin[e+f x]^2\right)^p dx$ when $\frac{m}{2} \in \mathbb{Z} \land p \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: $\cos[z]^2 == \frac{1}{1+\tan[z]^2}$

Basis: $\sin[z]^2 == \frac{\tan[z]^2}{1+\tan[z]^2}$

Basis: If $\frac{m}{2} \in \mathbb{Z}$, then $\cos[e+f x]^m F\left[\sin[e+f x]^2\right] == \frac{1}{f} \text{Subst}\left[\frac{F\left[\frac{x^2}{1+x^2}\right]}{\left(1+x^2\right)^{m/2+1}}, x, \tan[e+f x]\right] \partial_x \tan[e+f x]$

Rule: If $\frac{m}{2} \in \mathbb{Z} \land p \in \mathbb{Z}$, then

$$\int \cos[e+f x]^m \left(a+b \sin[e+f x]^2\right)^p dx \to \frac{1}{f} \text{Subst}\left[\int \frac{\left(a+(a+b) x^2\right)^p}{\left(1+x^2\right)^{m/2+p+1}} dx, x, \tan[e+f x]\right]$$

Program code:

```
Int[cos[e_.+f_.*x_]^m_*(a_+b_.*sin[e_.+f_.*x_]^2)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff/f*Subst[Int[(a+(a+b)*ff^2*x^2)^p/(1+ff^2*x^2)^(m/2+p+1),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,e,f},x] && IntegerQ[m/2] && IntegerQ[p]
```

2: $\int \cos[e+f x]^m \left(a+b \sin[e+f x]^2\right)^p dx$ when $\frac{m}{2} \in \mathbb{Z} \land p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{\cos[e+f x]^{m-1}}{\left(\cos[e+f x]^2\right)^{\frac{m-1}{2}}} == 0$

Basis: If $\frac{m}{2} \in \mathbb{Z}$, then $\frac{\text{Cos}[e+f x]^{m-1}}{\left(\text{Cos}[e+f x]^2\right)^{\frac{m-1}{2}}} == \frac{\sqrt{\text{Cos}[e+f x]^2}}{\text{Cos}[e+f x]}$

Basis: $\text{Cos}[e + f x]\ F[\text{Sin}[e + f x]] == \frac{1}{f}\ \text{Subst}[F[x],\ x,\ \text{Sin}[e + f x]]\ \partial_x \text{Sin}[e + f x]$

Rule: If $\frac{m}{2} \in \mathbb{Z}\ \land\ p \notin \mathbb{Z}$, then

$$\int \text{Cos}[e+f x]^m \left(a+b\,\text{Sin}[e+f x]^2\right)^p dx \longrightarrow \int \text{Cos}[e+f x]\ \text{Cos}[e+f x]^{m-1} \left(a+b\,\text{Sin}[e+f x]^2\right)^p dx$$

$$\longrightarrow \frac{\text{Cos}[e+f x]^{m-1}}{\left(\text{Cos}[e+f x]^2\right)^{\frac{m-1}{2}}} \int \text{Cos}[e+f x] \left(1-\text{Sin}[e+f x]^2\right)^{\frac{m-1}{2}} \left(a+b\,\text{Sin}[e+f x]^2\right)^p dx$$

$$\longrightarrow \frac{\sqrt{\text{Cos}[e+f x]^2}}{f\,\text{Cos}[e+f x]}\ \text{Subst}\left[\int \left(1-x^2\right)^{\frac{m-1}{2}} \left(a+b\,x^2\right)^p dx,\ x,\ \text{Sin}[e+f x]\right]$$

Program code:

```
Int[cos[e_.+f_.*x_]^m_*(a_+b_.*sin[e_.+f_.*x_]^2)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Sin[e+f*x],x]},
  ff*Sqrt[Cos[e+f*x]^2]/(f*Cos[e+f*x])*Subst[Int[(1-ff^2*x^2)^((m-1)/2)*(a+b*ff^2*x^2)^p,x],x,Sin[e+f*x]/ff]] /;
FreeQ[{a,b,e,f,p},x] && IntegerQ[m/2] && Not[IntegerQ[p]]
```

2: $\int (d\,\text{Cos}[e+f x])^m \left(a+b\,\text{Sin}[e+f x]^2\right)^p dx$ when $m \notin \mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{(d\,\text{Cos}[e+f x])^{m-1}}{\left(\text{Cos}[e+f x]^2\right)^{\frac{m-1}{2}}} == 0$

Basis: $\text{Cos}[e + f x]\ F[\text{Sin}[e + f x]] == \frac{1}{f}\ \text{Subst}[F[x],\ x,\ \text{Sin}[e + f x]]\ \partial_x \text{Sin}[e + f x]$

Rule:

$$\int (d \operatorname{Cos}[e+f x])^m \left(a+b \operatorname{Sin}[e+f x]^2\right)^p dx \longrightarrow d \int \operatorname{Cos}[e+f x] \, (d \operatorname{Cos}[e+f x])^{m-1} \left(a+b \operatorname{Sin}[e+f x]^2\right)^p dx$$

$$\longrightarrow \frac{d \, (d \operatorname{Cos}[e+f x])^{m-1}}{\left(\operatorname{Cos}[e+f x]^2\right)^{\frac{m-1}{2}}} \int \operatorname{Cos}[e+f x] \left(1-\operatorname{Sin}[e+f x]^2\right)^{\frac{m-1}{2}} \left(a+b \operatorname{Sin}[e+f x]^2\right)^p dx$$

$$\longrightarrow \frac{d^{2 \operatorname{IntPart}\left[\frac{m-1}{2}\right]+1} (d \operatorname{Cos}[e+f x])^{2 \operatorname{FracPart}\left[\frac{m-1}{2}\right]}}{f \left(\operatorname{Cos}[e+f x]^2\right)^{\operatorname{FracPart}\left[\frac{m-1}{2}\right]}} \operatorname{Subst}\left[\int \left(1-x^2\right)^{\frac{m-1}{2}} \left(a+b x^2\right)^p dx, x, \operatorname{Sin}[e+f x]\right]$$

Program code:

```
Int[(d_.*cos[e_.+f_.*x_])^m_*(a_+b_.*sin[e_.+f_.*x_]^2)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Sin[e+f*x],x]},
  ff*d^(2*IntPart[(m-1)/2]+1)*(d*Cos[e+f*x])^(2*FracPart[(m-1)/2])/(f*(Cos[e+f*x]^2)^FracPart[(m-1)/2])*
    Subst[Int[(1-ff^2*x^2)^((m-1)/2)*(a+b*ff^2*x^2)^p,x],x,Sin[e+f*x]/ff]] /;
FreeQ[{a,b,d,e,f,m,p},x] && Not[IntegerQ[m]]
```

5. $\int (d\,\text{Tan}[e+f\,x])^m \left(a+b\,\text{Sin}[e+f\,x]^2\right)^p\,dx$

1: $\int \text{Tan}[e+f\,x]^m \left(a+b\,\text{Sin}[e+f\,x]^2\right)^p\,dx$ when $\frac{m-1}{2}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $\text{Tan}[z]^2 == \frac{\text{Sin}[z]^2}{1-\text{Sin}[z]^2}$

Basis: If $\frac{m-1}{2}\in\mathbb{Z}$, then

$$\text{Tan}[e+f\,x]^m\,F\left[\text{Sin}[e+f\,x]^2\right] == \frac{1}{2\,f}\,\text{Subst}\left[\frac{x^{\frac{m-1}{2}}\,F[x]}{(1-x)^{\frac{m+1}{2}}},\,x,\,\text{Sin}[e+f\,x]^2\right]\,\partial_x\,\text{Sin}[e+f\,x]^2$$

Rule: If $\frac{m-1}{2}\in\mathbb{Z}$, then

$$\int \text{Tan}[e+f\,x]^m \left(a+b\,\text{Sin}[e+f\,x]^2\right)^p\,dx \rightarrow \frac{1}{2\,f}\,\text{Subst}\left[\int \frac{x^{\frac{m-1}{2}}\,(a+b\,x)^p}{(1-x)^{\frac{m+1}{2}}}\,dx,\,x,\,\text{Sin}[e+f\,x]^2\right]$$

Program code:

```
Int[tan[e_.+f_.*x_]^m_.*(a_+b_.*sin[e_.+f_.*x_]^2)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Sin[e+f*x]^2,x]},
  ff^((m+1)/2)/(2*f)*Subst[Int[x^((m-1)/2)*(a+b*ff*x)^p/(1-ff*x)^((m+1)/2),x],x,Sin[e+f*x]^2/ff]] /;
FreeQ[{a,b,e,f,p},x] && IntegerQ[(m-1)/2]
```

2: $\int (d\,\text{Tan}[e+f\,x])^m \left(a+b\,\text{Sin}[e+f\,x]^2\right)^p\,dx$ when $p\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $\text{Sin}[z]^2 == \frac{\text{Tan}[z]^2}{1+\text{Tan}[z]^2}$

Basis: $(d\,\mathrm{Tan}[e+f\,x])^m\,F\left[\mathrm{Sin}[e+f\,x]^2\right] = \frac{1}{f}\,\mathrm{Subst}\left[\frac{(d\,x)^m\,F\left[\frac{x^2}{1+x^2}\right]}{1+x^2},\,x,\,\mathrm{Tan}[e+f\,x]\right]\,\partial_x\mathrm{Tan}[e+f\,x]$

Rule: If $p\in\mathbb{Z}$, then

$$\int(d\,\mathrm{Tan}[e+f\,x])^m\,\left(a+b\,\mathrm{Sin}[e+f\,x]^n\right)^p\,dx \longrightarrow \frac{1}{f}\,\mathrm{Subst}\left[\int\frac{(d\,x)^m\,\left(a+(a+b)\,x^2\right)^p}{\left(1+x^2\right)^{p+1}}\,dx,\,x,\,\mathrm{Tan}[e+f\,x]\right]$$

Program code:

```
Int[(d_.*tan[e_.+f_.*x_])^m_*(a_+b_.*sin[e_.+f_.*x_]^2)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff/f*Subst[Int[(d*ff*x)^m*(a+(a+b)*ff^2*x^2)^p/(1+ff^2*x^2)^(p+1),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,d,e,f,m},x] && IntegerQ[p]
```

3: $\int\mathrm{Tan}[e+f\,x]^m\,\left(a+b\,\mathrm{Sin}[e+f\,x]^2\right)^p\,dx$ when $\frac{m}{2}\in\mathbb{Z}\,\land\,p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $\frac{m}{2}\in\mathbb{Z}$, then $\mathrm{Tan}[e+f\,x]^m = \frac{\mathrm{Sin}[e+f\,x]^m}{\left(\mathrm{Cos}[e+f\,x]^2\right)^{m/2}}$

Basis: $\partial_x\frac{\sqrt{\mathrm{Cos}[e+f\,x]^2}}{\mathrm{Cos}[e+f\,x]} = 0$

Basis: $\mathrm{Cos}[e+f\,x]\,F[\mathrm{Sin}[e+f\,x]] = \frac{1}{f}\,\mathrm{Subst}[F[x],\,x,\,\mathrm{Sin}[e+f\,x]]\,\partial_x\mathrm{Sin}[e+f\,x]$

Rule: If $\frac{m}{2}\in\mathbb{Z}\,\land\,p\notin\mathbb{Z}$, then

$$\int\mathrm{Tan}[e+f\,x]^m\,\left(a+b\,\mathrm{Sin}[e+f\,x]^2\right)^p\,dx \longrightarrow \int\frac{\mathrm{Sin}[e+f\,x]^m\,\left(a+b\,\mathrm{Sin}[e+f\,x]^2\right)^p}{\left(\mathrm{Cos}[e+f\,x]^2\right)^{m/2}}\,dx$$

$$\longrightarrow \frac{\sqrt{\mathrm{Cos}[e+f\,x]^2}}{\mathrm{Cos}[e+f\,x]}\int\frac{\mathrm{Cos}[e+f\,x]\,\mathrm{Sin}[e+f\,x]^m\,\left(a+b\,\mathrm{Sin}[e+f\,x]^2\right)^p}{\left(1-\mathrm{Sin}[e+f\,x]^2\right)^{\frac{m+1}{2}}}\,dx$$

$$\longrightarrow \frac{\sqrt{\text{Cos}[e+f\,x]^2}}{f\,\text{Cos}[e+f\,x]}\,\text{Subst}\left[\int\frac{x^m\,\left(a+b\,x^2\right)^p}{\left(1-x^2\right)^{\frac{m+1}{2}}}\,dx,\,x,\,\text{Sin}[e+f\,x]\right]$$

–

Program code:

```
Int[tan[e_.+f_.*x_]^m_*(a_+b_.*sin[e_.+f_.*x_]^2)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Sin[e+f*x],x]},
  ff^(m+1)*Sqrt[Cos[e+f*x]^2]/(f*Cos[e+f*x])*
    Subst[Int[x^m*(a+b*ff^2*x^2)^p/(1-ff^2*x^2)^((m+1)/2),x],x,Sin[e+f*x]/ff]] /;
FreeQ[{a,b,e,f,p},x] && IntegerQ[m/2] && Not[IntegerQ[p]]
```

4: $\int (d\,\mathrm{Tan}[e+f\,x])^m\,(a+b\,\mathrm{Sin}[e+f\,x]^2)^p\,dx$ when $m\notin\mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $\frac{m}{2}\in\mathbb{Z}$, then $\mathrm{Tan}[e+f\,x]^m == \frac{\mathrm{Sin}[e+f\,x]^m}{(\mathrm{Cos}[e+f\,x]^2)^{m/2}}$

Basis: $\partial_x \frac{(d\,\mathrm{Tan}[e+f\,x])^m\,(\mathrm{Cos}[e+f\,x]^2)^{m/2}}{\mathrm{Sin}[e+f\,x]^m} == 0$

Basis: $\mathrm{Cos}[e+f\,x]\,F[\mathrm{Sin}[e+f\,x]] == \frac{1}{f}\,\mathrm{Subst}[F[x],\,x,\,\mathrm{Sin}[e+f\,x]]\,\partial_x\mathrm{Sin}[e+f\,x]$

Rule: If $m\notin\mathbb{Z}$, then

$$\int (d\,\mathrm{Tan}[e+f\,x])^m\,(a+b\,\mathrm{Sin}[e+f\,x]^2)^p\,dx \to \frac{(d\,\mathrm{Tan}[e+f\,x])^m\,(\mathrm{Cos}[e+f\,x]^2)^{m/2}}{\mathrm{Sin}[e+f\,x]^m}\int\frac{\mathrm{Sin}[e+f\,x]^m\,(a+b\,\mathrm{Sin}[e+f\,x]^2)^p}{(\mathrm{Cos}[e+f\,x]^2)^{m/2}}\,dx$$

$$\to \frac{(d\,\mathrm{Tan}[e+f\,x])^{m+1}\,(\mathrm{Cos}[e+f\,x]^2)^{\frac{m+1}{2}}}{d\,\mathrm{Sin}[e+f\,x]^{m+1}}\int\frac{\mathrm{Cos}[e+f\,x]\,\mathrm{Sin}[e+f\,x]^m\,(a+b\,\mathrm{Sin}[e+f\,x]^2)^p}{(1-\mathrm{Sin}[e+f\,x]^2)^{\frac{m+1}{2}}}\,dx$$

$$\to \frac{(d\,\mathrm{Tan}[e+f\,x])^{m+1}\,(\mathrm{Cos}[e+f\,x]^2)^{\frac{m+1}{2}}}{d\,f\,\mathrm{Sin}[e+f\,x]^{m+1}}\,\mathrm{Subst}\left[\int\frac{x^m\,(a+b\,x^2)^p}{(1-x^2)^{\frac{m+1}{2}}}\,dx,\,x,\,\mathrm{Sin}[e+f\,x]\right]$$

Program code:

```
Int[(d_.*tan[e_.+f_.*x_])^m_*(a_+b_.*sin[e_.+f_.*x_]^2)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Sin[e+f*x],x]},
  ff*(d*Tan[e+f*x])^(m+1)*(Cos[e+f*x]^2)^((m+1)/2)/(d*f*Sin[e+f*x]^(m+1))*
    Subst[Int[(ff*x)^m*(a+b*ff^2*x^2)^p/(1-ff^2*x^2)^((m+1)/2),x],x,Sin[e+f*x]/ff]] /;
FreeQ[{a,b,d,e,f,m,p},x] && Not[IntegerQ[m]]
```

6. $\int (c \operatorname{Cos}[e+f x])^m (d \operatorname{Sin}[e+f x])^n \left(a+b \operatorname{Sin}[e+f x]^2\right)^p dx$

1: $\int \operatorname{Cos}[e+f x]^m (d \operatorname{Sin}[e+f x])^n \left(a+b \operatorname{Sin}[e+f x]^2\right)^p dx$ when $\frac{m-1}{2} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{m-1}{2} \in \mathbb{Z}$, then

$$\operatorname{Cos}[e+f x]^m F[\operatorname{Sin}[e+f x]] = \frac{1}{f} \operatorname{Subst}\left[\left(1-x^2\right)^{\frac{m-1}{2}} F[x],\, x,\, \operatorname{Sin}[e+f x]\right] \partial_x \operatorname{Sin}[e+f x]$$

Rule: If $\frac{m-1}{2} \in \mathbb{Z}$, then

$$\int \operatorname{Cos}[e+f x]^m (d \operatorname{Sin}[e+f x])^n \left(a+b \operatorname{Sin}[e+f x]^2\right)^p dx \to \frac{1}{f} \operatorname{Subst}\left[\int (d\, x)^n \left(1-x^2\right)^{\frac{m-1}{2}} \left(a+b x^2\right)^p dx,\, x,\, \operatorname{Sin}[e+f x]\right]$$

Program code:

```
Int[cos[e_.+f_.*x_]^m_.*(d_.*sin[e_.+f_.*x_])^n_.*(a_+b_.*sin[e_.+f_.*x_]^2)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Sin[e+f*x],x]},
  ff/f*Subst[Int[(d*ff*x)^n*(1-ff^2*x^2)^((m-1)/2)*(a+b*ff^2*x^2)^p,x],x,Sin[e+f*x]/ff]] /;
FreeQ[{a,b,d,e,f,n,p},x] && IntegerQ[(m-1)/2]
```

2: $\int (c \operatorname{Cos}[e+f x])^m \operatorname{Sin}[e+f x]^n \left(a+b \operatorname{Sin}[e+f x]^2\right)^p dx$ when $\frac{n-1}{2} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: $\operatorname{Sin}[z]^2 == 1-\operatorname{Cos}[z]^2$

Basis: If $\frac{n-1}{2} \in \mathbb{Z}$, then

$$\operatorname{Sin}[e+f x]^n F\left[\operatorname{Sin}[e+f x]^2\right] == -\frac{1}{f} \operatorname{Subst}\left[\left(1-x^2\right)^{\frac{n-1}{2}} F\left[1-x^2\right],\ x,\ \operatorname{Cos}[e+f x]\right] \partial_x \operatorname{Cos}[e+f x]$$

Rule: If $\frac{n-1}{2} \in \mathbb{Z}$, then

$$\int (c \operatorname{Cos}[e+f x])^m \operatorname{Sin}[e+f x]^n \left(a+b \operatorname{Sin}[e+f x]^2\right)^p dx \rightarrow -\frac{1}{f} \operatorname{Subst}\left[\int (c x)^m \left(1-x^2\right)^{\frac{n-1}{2}} \left(a+b-b x^2\right)^p dx,\ x,\ \operatorname{Cos}[e+f x]\right]$$

Program code:

```
Int[(c_.*sin[e_.+f_.*x_])^m_*sin[e_.+f_.*x_]^n_.*(a_+b_.*sin[e_.+f_.*x_]^2)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Cos[e+f*x],x]},
  -ff/f*Subst[Int[(c*ff*x)^m*(1-ff^2*x^2)^((n-1)/2)*(a+b-b*ff^2*x^2)^p,x],x,Cos[e+f*x]/ff]] /;
FreeQ[{a,b,c,e,f,m,p},x] && IntegerQ[(n-1)/2]
```

3. $\int (c \operatorname{Cos}[e+f x])^m (d \operatorname{Sin}[e+f x])^n \left(a+b \operatorname{Sin}[e+f x]^2\right)^p dx$ when $\frac{m}{2} \in \mathbb{Z}$

1: $\int \operatorname{Cos}[e+f x]^m \operatorname{Sin}[e+f x]^n \left(a+b \operatorname{Sin}[e+f x]^2\right)^p dx$ when $\frac{m}{2} \in \mathbb{Z} \land \frac{n}{2} \in \mathbb{Z} \land p \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: $\operatorname{Cos}[z]^2 == \frac{1}{1+\operatorname{Tan}[z]^2}$

Basis: $\operatorname{Sin}[z]^2 == \frac{\operatorname{Tan}[z]^2}{1+\operatorname{Tan}[z]^2}$

Basis: If $\frac{m}{2} \in \mathbb{Z}$, then $\text{Cos}[e+f x]^m F\left[\text{Sin}[e+f x]^2\right] == \frac{1}{f} \text{Subst}\left[\frac{F\left[\frac{x^2}{1+x^2}\right]}{\left(1+x^2\right)^{m/2+1}}, x, \text{Tan}[e+f x]\right] \partial_x \text{Tan}[e+f x]$

Rule: If $\frac{m}{2} \in \mathbb{Z} \land \frac{n}{2} \in \mathbb{Z} \land p \in \mathbb{Z}$, then

$$\int \text{Cos}[e+f x]^m \text{Sin}[e+f x]^n \left(a+b \text{Sin}[e+f x]^2\right)^p dx \to \frac{1}{f} \text{Subst}\left[\int \frac{x^n \left(a+(a+b) x^2\right)^p}{\left(1+x^2\right)^{(m+n)/2+p+1}} dx, x, \text{Tan}[e+f x]\right]$$

Program code:

```
Int[cos[e_.+f_.*x_]^m_*sin[e_.+f_.*x_]^n_*(a_+b_.*sin[e_.+f_.*x_]^2)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff^(n+1)/f*Subst[Int[x^n*(a+(a+b)*ff^2*x^2)^p/(1+ff^2*x^2)^((m+n)/2+p+1),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,e,f},x] && IntegerQ[m/2] && IntegerQ[n/2] && IntegerQ[p]
```

2: $\int \text{Cos}[e+f x]^m (d \text{Sin}[e+f x])^n \left(a+b \text{Sin}[e+f x]^2\right)^p dx$ when $\frac{m}{2} \in \mathbb{Z} \land \neg\left(\frac{n}{2} \in \mathbb{Z} \land p \in \mathbb{Z}\right)$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{\text{Cos}[e+f x]^{m-1}}{\left(\text{Cos}[e+f x]^2\right)^{\frac{m-1}{2}}} == 0$

Basis: If $\frac{m}{2} \in \mathbb{Z}$, then $\frac{\text{Cos}[e+f x]^{m-1}}{\left(\text{Cos}[e+f x]^2\right)^{\frac{m-1}{2}}} == \frac{\sqrt{\text{Cos}[e+f x]^2}}{\text{Cos}[e+f x]}$

Basis: $\text{Cos}[e+f x] F[\text{Sin}[e+f x]] == \frac{1}{f} \text{Subst}[F[x], x, \text{Sin}[e+f x]] \partial_x \text{Sin}[e+f x]$

Rule: If $\frac{m}{2} \in \mathbb{Z}$, then

$$\int \text{Cos}[e+f x]^m (d \text{Sin}[e+f x])^n \left(a+b \text{Sin}[e+f x]^2\right)^p dx \to \int \text{Cos}[e+f x] \text{Cos}[e+f x]^{m-1} (d \text{Sin}[e+f x])^n \left(a+b \text{Sin}[e+f x]^2\right)^p dx$$

$$\to \frac{\text{Cos}[e+f x]^{m-1}}{\left(\text{Cos}[e+f x]^2\right)^{\frac{m-1}{2}}} \int \text{Cos}[e+f x] \left(1-\text{Sin}[e+f x]^2\right)^{\frac{m-1}{2}} (d \text{Sin}[e+f x])^n \left(a+b \text{Sin}[e+f x]^2\right)^p dx$$

$$\longrightarrow \frac{\sqrt{\text{Cos}[e+f\,x]^2}}{f\,\text{Cos}[e+f\,x]}\,\text{Subst}\left[\int (d\,x)^n\,\left(1-x^2\right)^{\frac{m-1}{2}}\,\left(a+b\,x^2\right)^p\,dx,\,x,\,\text{Sin}[e+f\,x]\right]$$

–

Program code:

```
Int[cos[e_.+f_.*x_]^m_*(d_.*sin[e_.+f_.*x_])^n_.*(a_+b_.*sin[e_.+f_.*x_]^2)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Sin[e+f*x],x]},
  ff*Sqrt[Cos[e+f*x]^2]/(f*Cos[e+f*x])*Subst[Int[(d*ff*x)^n*(1-ff^2*x^2)^((m-1)/2)*(a+b*ff^2*x^2)^p,x],x,Sin[e+f*x]/ff]] /;
FreeQ[{a,b,d,e,f,n,p},x] && IntegerQ[m/2]
```

4: $\int (c\,\mathrm{Cos}[e+f\,x])^m\,(d\,\mathrm{Sin}[e+f\,x])^n\,\left(a+b\,\mathrm{Sin}[e+f\,x]^2\right)^p\,dx$ when $m\notin\mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{(c\,\mathrm{Cos}[e+f\,x])^{m-1}}{\left(\mathrm{Cos}[e+f\,x]^2\right)^{\frac{m-1}{2}}} == 0$

Basis: $\mathrm{Cos}[e+f\,x]\,F[\mathrm{Sin}[e+f\,x]] == \frac{1}{f}\,\mathrm{Subst}[F[x],\,x,\,\mathrm{Sin}[e+f\,x]]\,\partial_x \mathrm{Sin}[e+f\,x]$

Rule:

$$\int (c\,\mathrm{Cos}[e+f\,x])^m\,(d\,\mathrm{Sin}[e+f\,x])^n\,\left(a+b\,\mathrm{Sin}[e+f\,x]^2\right)^p\,dx \to c\int \mathrm{Cos}[e+f\,x]\,(c\,\mathrm{Cos}[e+f\,x])^{m-1}\,(d\,\mathrm{Sin}[e+f\,x])^n\,\left(a+b\,\mathrm{Sin}[e+f\,x]^2\right)^p\,dx$$

$$\to \frac{c\,(c\,\mathrm{Cos}[e+f\,x])^{m-1}}{\left(\mathrm{Cos}[e+f\,x]^2\right)^{\frac{m-1}{2}}}\int \mathrm{Cos}[e+f\,x]\,\left(1-\mathrm{Sin}[e+f\,x]^2\right)^{\frac{m-1}{2}}\,(d\,\mathrm{Sin}[e+f\,x])^n\,\left(a+b\,\mathrm{Sin}[e+f\,x]^2\right)^p\,dx$$

$$\to \frac{c^{2\,\mathrm{IntPart}\left[\frac{m-1}{2}\right]+1}\,(c\,\mathrm{Cos}[e+f\,x])^{2\,\mathrm{FracPart}\left[\frac{m-1}{2}\right]}}{f\,\left(\mathrm{Cos}[e+f\,x]^2\right)^{\mathrm{FracPart}\left[\frac{m-1}{2}\right]}}\,\mathrm{Subst}\left[\int (d\,x)^n\,\left(1-x^2\right)^{\frac{m-1}{2}}\,\left(a+b\,x^2\right)^p\,dx,\,x,\,\mathrm{Sin}[e+f\,x]\right]$$

Program code:

```
Int[(c_.*cos[e_.+f_.*x_])^m_*(d_.*sin[e_.+f_.*x_])^n_.*(a_+b_.*sin[e_.+f_.*x_]^2)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Sin[e+f*x],x]},
  ff*c^(2*IntPart[(m-1)/2]+1)*(c*Cos[e+f*x])^(2*FracPart[(m-1)/2])/(f*(Cos[e+f*x]^2)^FracPart[(m-1)/2])*
    Subst[Int[(d*ff*x)^n*(1-ff^2*x^2)^((m-1)/2)*(a+b*ff^2*x^2)^p,x],x,Sin[e+f*x]/ff]] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && Not[IntegerQ[m]]
```

Rules for integrands of the form $(d\,\text{Trig}[e+f\,x])^m\,(a+b\,(c\,\text{Sin}[e+f\,x])^n)^p$

1. $\int (d\,\text{Trig}[e+f\,x])^m\,(b\,(c\,\text{Sin}[e+f\,x])^n)^p\,dx$ when $p\notin\mathbb{Z}$

   1. $\int (b\,\text{Sin}[e+f\,x]^2)^p\,dx$ when $p\notin\mathbb{Z}$

      1: $\int (b\,\text{Sin}[e+f\,x]^2)^p\,dx$ when $p\notin\mathbb{Z}\,\land\,p>1$

Rule: If $p\notin\mathbb{Z}\,\land\,p>1$, then

$$\int (b\,\text{Sin}[e+f\,x]^2)^p\,dx \longrightarrow -\frac{\text{Cot}[e+f\,x]\,(b\,\text{Sin}[e+f\,x]^2)^p}{2\,f\,p}+\frac{b\,(2\,p-1)}{2\,p}\int (b\,\text{Sin}[e+f\,x]^2)^{p-1}\,dx$$

Program code:

```
Int[(b_.*sin[e_.+f_.*x_]^2)^p_,x_Symbol] :=
  -Cot[e+f*x]*(b*Sin[e+f*x]^2)^p/(2*f*p) +
  b*(2*p-1)/(2*p)*Int[(b*Sin[e+f*x]^2)^(p-1),x] /;
FreeQ[{b,e,f},x] && Not[IntegerQ[p]] && GtQ[p,1]
```

2: $\int \left(b \operatorname{Sin}[e+f x]^2\right)^p \, dx$ when $p \notin \mathbb{Z} \land p < -1$

Rule: If $p \notin \mathbb{Z} \land p < -1$, then

$$\int \left(b \operatorname{Sin}[e+f x]^2\right)^p \, dx \longrightarrow \frac{\operatorname{Cot}[e+f x] \left(b \operatorname{Sin}[e+f x]^2\right)^{p+1}}{b f (2 p+1)} + \frac{2 (p+1)}{b (2 p+1)} \int \left(b \operatorname{Sin}[e+f x]^2\right)^{p+1} dx$$

Program code:

```
Int[(b_.*sin[e_.+f_.*x_]^2)^p_,x_Symbol] :=
  Cot[e+f*x]*(b*Sin[e+f*x]^2)^(p+1)/(b*f*(2*p+1)) +
  2*(p+1)/(b*(2*p+1))*Int[(b*Sin[e+f*x]^2)^(p+1),x] /;
FreeQ[{b,e,f},x] && Not[IntegerQ[p]] && LtQ[p,-1]
```

2. $\int \mathrm{Tan}[e+f x]^m \left(b \left(c \,\mathrm{Sin}[e+f x]\right)^n\right)^p \, dx$ when $\frac{m-1}{2} \in \mathbb{Z}$

1: $\int \mathrm{Tan}[e+f x]^m \left(b \,\mathrm{Sin}[e+f x]^n\right)^p \, dx$ when $\frac{m-1}{2} \in \mathbb{Z} \land \frac{n}{2} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: $\mathrm{Tan}[z]^2 == \frac{\mathrm{Sin}[z]^2}{1-\mathrm{Sin}[z]^2}$

Basis: If $\frac{m-1}{2} \in \mathbb{Z}$, then

$\mathrm{Tan}[e+f x]^m F\left[\mathrm{Sin}[e+f x]^2\right] == \frac{1}{2 f} \mathrm{Subst}\left[\frac{x^{\frac{m-1}{2}} F[x]}{(1-x)^{\frac{m+1}{2}}}, x, \mathrm{Sin}[e+f x]^2\right] \partial_x \mathrm{Sin}[e+f x]^2$

Rule: If $\frac{m-1}{2} \in \mathbb{Z} \land \frac{n}{2} \in \mathbb{Z}$, then

$$\int \mathrm{Tan}[e+f x]^m \left(b \,\mathrm{Sin}[e+f x]^n\right)^p \, dx \to \frac{1}{2 f} \mathrm{Subst}\left[\int \frac{x^{\frac{m-1}{2}} \left(b x^{n/2}\right)^p}{(1-x)^{\frac{m+1}{2}}} \, dx, x, \mathrm{Sin}[e+f x]^2\right]$$

Program code:

```
Int[tan[e_.+f_.*x_]^m_.*(b_.*sin[e_.+f_.*x_]^n_)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Sin[e+f*x]^2,x]},
  ff^((m+1)/2)/(2*f)*Subst[Int[x^((m-1)/2)*(b*ff^(n/2)*x^(n/2))^p/(1-ff*x)^((m+1)/2),x],x,Sin[e+f*x]^2/ff]] /;
FreeQ[{b,e,f,p},x] && IntegerQ[(m-1)/2] && IntegerQ[n/2]
```

2: $\int \mathrm{Tan}[e+f x]^m \left(b \left(c \,\mathrm{Sin}[e+f x]\right)^n\right)^p \, dx$ when $\frac{m-1}{2} \in \mathbb{Z}^-$

Derivation: Integration by substitution

Basis: $\mathrm{Tan}[z]^2 == \frac{\mathrm{Sin}[z]^2}{1-\mathrm{Sin}[z]^2}$

Basis: If $\frac{m-1}{2} \in \mathbb{Z}$, then $\text{Tan}[e+f x]^m F[\text{Sin}[e+f x]] == \frac{1}{f} \text{Subst}\left[\frac{x^m F[x]}{\left(1-x^2\right)^{\frac{m+1}{2}}}, x, \text{Sin}[e+f x]\right] \partial_x \text{Sin}[e+f x]$

Rule: If $\frac{m-1}{2} \in \mathbb{Z}^-$, then

$$\int \text{Tan}[e+f x]^m \left(b \left(c \,\text{Sin}[e+f x]\right)^n\right)^p dx \longrightarrow \frac{1}{f} \text{Subst}\left[\int \frac{x^m \left(b \left(c x\right)^n\right)^p}{\left(1-x^2\right)^{\frac{m+1}{2}}} dx, x, \text{Sin}[e+f x]\right]$$

Program code:

```
Int[tan[e_.+f_.*x_]^m_.*(b_.*(c_.*sin[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Sin[e+f*x],x]},
  ff^(m+1)/f*Subst[Int[x^m*(b*(c*ff*x)^n)^p/(1-ff^2*x^2)^((m+1)/2),x],x,Sin[e+f*x]/ff]] /;
FreeQ[{b,c,e,f,n,p},x] && ILtQ[(m-1)/2,0]
```

3: $\int u \left(b \operatorname{Sin}[e+f x]^n\right)^p \, dx$ when $p \notin \mathbb{Z} \land n \in \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(b \operatorname{Sin}[e+f x]^n)^p}{\operatorname{Sin}[e+f x]^{n p}} == 0$

Rule: If $p \notin \mathbb{Z} \land n \in \mathbb{Z}$, then

$$\int u \left(b \operatorname{Sin}[e+f x]^n\right)^p dx \longrightarrow \frac{b^{\operatorname{IntPart}[p]} \left(b \operatorname{Sin}[e+f x]^n\right)^{\operatorname{FracPart}[p]}}{\operatorname{Sin}[e+f x]^{n \operatorname{FracPart}[p]}} \int u \operatorname{Sin}[e+f x]^{n p} dx$$

Program code:

```
Int[u_.*(b_.*sin[e_.+f_.*x_]^n_)^p_,x_Symbol] :=
  With[{ff=FreeFactors[Sin[e+f*x],x]},
  (b*ff^n)^IntPart[p]*(b*Sin[e+f*x]^n)^FracPart[p]/(Sin[e+f*x]/ff)^(n*FracPart[p])*
    Int[ActivateTrig[u]*(Sin[e+f*x]/ff)^(n*p),x]] /;
FreeQ[{b,e,f,n,p},x] && Not[IntegerQ[p]] && IntegerQ[n] &&
  (EqQ[u,1] || MatchQ[u,(d_.*trig_[e+f*x])^m_. /; FreeQ[{d,m},x] && MemberQ[{sin,cos,tan,cot,sec,csc},trig]])
```

4: $\int u\,(b\,(c\,\mathrm{Sin}[e+f\,x])^n)^p\,dx$ when $p\notin\mathbb{Z}\wedge n\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(b\,(c\,\mathrm{Sin}[e+f\,x])^n)^p}{(c\,\mathrm{Sin}[e+f\,x])^{n\,p}} == 0$

Rule: If $p\notin\mathbb{Z}\wedge n\notin\mathbb{Z}$, then

$$\int u\,(b\,(c\,\mathrm{Sin}[e+f\,x])^n)^p\,dx \to \frac{b^{\mathrm{IntPart}[p]}\,(b\,(c\,\mathrm{Sin}[e+f\,x])^n)^{\mathrm{FracPart}[p]}}{(c\,\mathrm{Sin}[e+f\,x])^{n\,\mathrm{FracPart}[p]}}\int u\,(c\,\mathrm{Sin}[e+f\,x])^{n\,p}\,dx$$

Program code:

```
Int[u_.*(b_.*(c_.*sin[e_.+f_.*x_])^n_)^p_,x_Symbol] :=
  b^IntPart[p]*(b*(c*Sin[e+f*x])^n)^FracPart[p]/(c*Sin[e+f*x])^(n*FracPart[p])*
    Int[ActivateTrig[u]*(c*Sin[e+f*x])^(n*p),x] /;
FreeQ[{b,c,e,f,n,p},x] && Not[IntegerQ[p]] && Not[IntegerQ[n]] &&
  (EqQ[u,1] || MatchQ[u,(d_.*trig_[e+f*x])^m_. /; FreeQ[{d,m},x] && MemberQ[{sin,cos,tan,cot,sec,csc},trig]])
```

2. $\int (a+b\,(c\,\mathrm{Sin}[e+f\,x])^n)^p\,dx$

1. $\int (a+b\,\mathrm{Sin}[e+f\,x]^4)^p\,dx$

x: $\int (a+b\,\mathrm{Sin}[e+f\,x]^4)^p\,dx$ when $p\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $\mathrm{Sin}[z]^2 == \frac{1}{1+\mathrm{Cot}[z]^2}$

Basis: $F\left[\mathrm{Sin}[e+f\,x]^2\right] == -\frac{1}{f}\,\mathrm{Subst}\left[\frac{F\left[\frac{1}{1+x^2}\right]}{1+x^2},\ x,\ \mathrm{Cot}[e+f\,x]\right]\,\partial_x\mathrm{Cot}[e+f\,x]$

Rule: If $p\in\mathbb{Z}$, then

$$\int (a+b\,\mathrm{Sin}[e+f\,x]^4)^p\,dx \to -\frac{1}{f}\,\mathrm{Subst}\left[\int\frac{(a+b+2\,a\,x^2+a\,x^4)^p}{(1+x^2)^{2\,p+1}}\,dx,\ x,\ \mathrm{Cot}[e+f\,x]\right]$$

Program code:

```
(* Int[(a_+b_.*sin[e_.+f_.*x_]^4)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  -ff/f*Subst[Int[(a+b+2*a*ff^2*x^2+a*ff^4*x^4)^p/(1+ff^2*x^2)^(2*p+1),x],x,Cot[e+f*x]/ff]] /;
FreeQ[{a,b,e,f},x] && IntegerQ[p] *)
```

1: $\int (a+b\,\mathrm{Sin}[e+f\,x]^4)^p\,dx$ when $p\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $\mathrm{Sin}[z]^2 == \frac{\mathrm{Tan}[z]^2}{1+\mathrm{Tan}[z]^2}$

Basis: $F\left[\operatorname{Sin}[e+f x]^2\right] = \frac{1}{f}\operatorname{Subst}\left[\frac{F\left[\frac{x^2}{1+x^2}\right]}{1+x^2}, x, \operatorname{Tan}[e+f x]\right] \partial_x \operatorname{Tan}[e+f x]$

Rule: If $p \in \mathbb{Z}$, then

$$\int \left(a+b \operatorname{Sin}[e+f x]^4\right)^p dx \longrightarrow \frac{1}{f}\operatorname{Subst}\left[\int \frac{\left(a+2 a x^2+(a+b) x^4\right)^p}{\left(1+x^2\right)^{2 p+1}} dx, x, \operatorname{Tan}[e+f x]\right]$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_]^4)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff/f*Subst[Int[(a+2*a*ff^2*x^2+(a+b)*ff^4*x^4)^p/(1+ff^2*x^2)^(2*p+1),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,e,f},x] && IntegerQ[p]
```

2: $\int \left(a+b\,\text{Sin}[e+f\,x]^4\right)^p\,dx$ when $p-\frac{1}{2}\in\mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $a+b\,\text{Sin}[z]^4 = \frac{a+2\,a\,\text{Tan}[z]^2+(a+b)\,\text{Tan}[z]^4}{\text{Sec}[z]^4}$

Basis: $\partial_x \frac{\left(a+b\,\text{Sin}[e+f\,x]^4\right)^p\,\left(\text{Sec}[e+f\,x]^2\right)^{2p}}{\left(a+2\,a\,\text{Tan}[e+f\,x]^2+(a+b)\,\text{Tan}[e+f\,x]^4\right)^p} = 0$

Basis: $F[\text{Tan}[e+f\,x]] = \frac{1}{f}\,\text{Subst}\left[\frac{F[x]}{1+x^2},\,x,\,\text{Tan}[e+f\,x]\right]\partial_x \text{Tan}[e+f\,x]$

Rule: If $p-\frac{1}{2}\in\mathbb{Z}$, then

$$\int \left(a+b\,\text{Sin}[e+f\,x]^4\right)^p\,dx \to \frac{\left(a+b\,\text{Sin}[e+f\,x]^4\right)^p\,\left(\text{Sec}[e+f\,x]^2\right)^{2p}}{\left(a+2\,a\,\text{Tan}[e+f\,x]^2+(a+b)\,\text{Tan}[e+f\,x]^4\right)^p}\int \frac{\left(a+2\,a\,\text{Tan}[e+f\,x]^2+(a+b)\,\text{Tan}[e+f\,x]^4\right)^p}{\left(1+\text{Tan}[e+f\,x]^2\right)^{2p}}\,dx$$

$$\to \frac{\left(a+b\,\text{Sin}[e+f\,x]^4\right)^p\,\left(\text{Sec}[e+f\,x]^2\right)^{2p}}{f\,\left(a+2\,a\,\text{Tan}[e+f\,x]^2+(a+b)\,\text{Tan}[e+f\,x]^4\right)^p}\,\text{Subst}\left[\int \frac{\left(a+2\,a\,x^2+(a+b)\,x^4\right)^p}{\left(1+x^2\right)^{2p+1}}\,dx,\,x,\,\text{Tan}[e+f\,x]\right]$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_]^4)^p_,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff*(a+b*Sin[e+f*x]^4)^p*(Sec[e+f*x]^2)^(2*p)/(f*(a+2*a*Tan[e+f*x]^2+(a+b)*Tan[e+f*x]^4)^p)*
    Subst[Int[(a+2*a*ff^2*x^2+(a+b)*ff^4*x^4)^p/(1+ff^2*x^2)^(2*p+1),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,e,f,p},x] && IntegerQ[p-1/2]
```

2: $\int \frac{1}{a+b\,\text{Sin}[e+f\,x]^n}\,dx$ when $\frac{n}{2}\in\mathbb{Z}$

Derivation: Algebraic expansion

Basis: If $\frac{n}{2} \in \mathbb{Z}^+$, then $\frac{1}{a+b z^n} == \frac{2}{a n} \sum_{k=1}^{n/2} \frac{1}{1-(-1)^{-4 k/n} \left(-\frac{a}{b}\right)^{-2/n} z^2}$

Rule: If $\frac{n}{2} \in \mathbb{Z}$, then

$$\int \frac{1}{a+b \operatorname{Sin}[e+f x]^n} \, dx \to \frac{2}{a n} \sum_{k=1}^{n/2} \int \frac{1}{1-(-1)^{-4 k/n} \left(-\frac{a}{b}\right)^{-2/n} \operatorname{Sin}[e+f x]^2} \, dx$$

Program code:

```
Int[1/(a_+b_.*sin[e_.+f_.*x_]^n_),x_Symbol] :=
  Module[{k},
  Dist[2/(a*n),Sum[Int[1/(1-Sin[e+f*x]^2/((-1)^(4*k/n)*Rt[-a/b,n/2])),x],{k,1,n/2}],x]] /;
FreeQ[{a,b,e,f},x] && IntegerQ[n/2]
```

x: $\int (a+b \operatorname{Sin}[e+f x]^n)^p \, dx$ when $\frac{n}{2} \in \mathbb{Z} \land p \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $F\left[\operatorname{Sin}[e+f x]^2\right] == -\frac{1}{f} \operatorname{Subst}\left[\frac{F\left[\frac{1}{1+x^2}\right]}{1+x^2}, x, \operatorname{Cot}[e+f x]\right] \partial_x \operatorname{Cot}[e+f x]$

–

Rule: If $\frac{n}{2} \in \mathbb{Z} \land p \in \mathbb{Z}^+$, then

$$\int (a+b \operatorname{Sin}[e+f x]^n)^p \, dx \to -\frac{1}{f} \operatorname{Subst}\left[\int \frac{\left(b+a \left(1+x^2\right)^{n/2}\right)^p}{\left(1+x^2\right)^{n p/2+1}} \, dx, x, \operatorname{Cot}[e+f x]\right]$$

–

Program code:

```
(* Int[(a_+b_.*sin[e_.+f_.*x_]^n_)^p_,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  -ff/f*Subst[Int[(b+a*(1+ff^2*x^2)^(n/2))^p/(1+ff^2*x^2)^(n*p/2+1),x],x,Cot[e+f*x]/ff]] /;
FreeQ[{a,b,e,f},x] && IntegerQ[n/2] && IGtQ[p,0] *)
```

3: $\int (a + b \operatorname{Sin}[e + f x]^n)^p \, dx$ when $\frac{n}{2} \in \mathbb{Z} \land p \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $F\left[\operatorname{Sin}[e + f x]^2\right] == \frac{1}{f} \operatorname{Subst}\left[\frac{F\left[\frac{x^2}{1+x^2}\right]}{1+x^2}, x, \operatorname{Tan}[e + f x]\right] \partial_x \operatorname{Tan}[e + f x]$

Rule: If $\frac{n}{2} \in \mathbb{Z} \land p \in \mathbb{Z}^+$, then

$$\int (a + b \operatorname{Sin}[e + f x]^n)^p \, dx \to \frac{1}{f} \operatorname{Subst}\left[\int \frac{\left(b x^n + a \left(1 + x^2\right)^{n/2}\right)^p}{\left(1 + x^2\right)^{n p/2+1}} \, dx, x, \operatorname{Tan}[e + f x]\right]$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_]^n_)^p_,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff/f*Subst[Int[(b*ff^n*x^n+a*(1+ff^2*x^2)^(n/2))^p/(1+ff^2*x^2)^(n*p/2+1),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,e,f},x] && IntegerQ[n/2] && IGtQ[p,0]
```

4: $\int (a + b (c \operatorname{Sin}[e + f x])^n)^p \, dx$ when $p \in \mathbb{Z}^+ \lor (p == -1 \land n \in \mathbb{Z})$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}^+ \lor (p == -1 \land n \in \mathbb{Z})$, then

$$\int (a + b (c \operatorname{Sin}[e + f x])^n)^p \, dx \longrightarrow \int \operatorname{ExpandTrig}[(a + b (c \operatorname{Sin}[e + f x])^n)^p, x] \, dx$$

Program code:

```
Int[(a_+b_.*(c_.*sin[e_.+f_.*x_])^n_)^p_,x_Symbol] :=
  Int[ExpandTrig[(a+b*(c*sin[e+f*x])^n)^p,x],x] /;
FreeQ[{a,b,c,e,f,n},x] && (IGtQ[p,0] || EqQ[p,-1] && IntegerQ[n])
```

U: $\int (a + b (c \operatorname{Sin}[e + f x])^n)^p \, dx$

Rule:

$$\int (a + b (c \operatorname{Sin}[e + f x])^n)^p \, dx \longrightarrow \int (a + b (c \operatorname{Sin}[e + f x])^n)^p \, dx$$

Program code:

```
Int[(a_+b_.*(c_.*sin[e_.+f_.*x_])^n_)^p_,x_Symbol] :=
  Unintegrable[(a+b*(c*Sin[e+f*x])^n)^p,x] /;
FreeQ[{a,b,c,e,f,n,p},x]
```

3. $\int (d\,\text{Sin}[e+f\,x])^m\,(a+b\,(c\,\text{Sin}[e+f\,x])^n)^p\,dx$

1: $\int \text{Sin}[e+f\,x]^m\,(a+b\,\text{Sin}[e+f\,x]^n)^p\,dx$ when $\frac{m-1}{2}\in\mathbb{Z}\,\land\,\frac{n}{2}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $\text{Sin}[z]^2 == 1-\text{Cos}[z]^2$

Basis: If $\frac{m-1}{2}\in\mathbb{Z}$, then

$$\text{Sin}[e+f\,x]^m\,F\left[\text{Sin}[e+f\,x]^2\right] == -\frac{1}{f}\,\text{Subst}\left[\left(1-x^2\right)^{\frac{m-1}{2}}\,F\left[1-x^2\right],\,x,\,\text{Cos}[e+f\,x]\right]\,\partial_x \text{Cos}[e+f\,x]$$

Rule: If $\frac{m-1}{2}\in\mathbb{Z}\,\land\,\frac{n}{2}\in\mathbb{Z}$, then

$$\int \text{Sin}[e+f\,x]^m\,(a+b\,\text{Sin}[e+f\,x]^n)^p\,dx \longrightarrow -\frac{1}{f}\,\text{Subst}\left[\int \left(1-x^2\right)^{\frac{m-1}{2}}\,\left(a+b\,\left(1-x^2\right)^{n/2}\right)^p\,dx,\,x,\,\text{Cos}[e+f\,x]\right]$$

Program code:

```
Int[sin[e_.+f_.*x_]^m_.*(a_+b_.*sin[e_.+f_.*x_]^4)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Cos[e+f*x],x]},
  -ff/f*Subst[Int[(1-ff^2*x^2)^((m-1)/2)*(a+b-2*b*ff^2*x^2+b*ff^4*x^4)^p,x],x,Cos[e+f*x]/ff]] /;
FreeQ[{a,b,e,f,p},x] && IntegerQ[(m-1)/2]
```

```
Int[sin[e_.+f_.*x_]^m_.*(a_+b_.*sin[e_.+f_.*x_]^n_)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Cos[e+f*x],x]},
  -ff/f*Subst[Int[(1-ff^2*x^2)^((m-1)/2)*(a+b*(1-ff^2*x^2)^(n/2))^p,x],x,Cos[e+f*x]/ff]] /;
FreeQ[{a,b,e,f,p},x] && IntegerQ[(m-1)/2] && IntegerQ[n/2]
```

2: $\int \text{Sin}[e+fx]^m \left(a+b\,\text{Sin}[e+fx]^n\right)^p dx$ when $\frac{m}{2}\in\mathbb{Z}\land\frac{n}{2}\in\mathbb{Z}\land p\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $\text{Sin}[z]^2 == \frac{\text{Tan}[z]^2}{1+\text{Tan}[z]^2}$

Basis: If $\frac{m}{2}\in\mathbb{Z}$, then $\text{Sin}[e+fx]^m\,F\left[\text{Sin}[e+fx]^2\right] == \frac{1}{f}\,\text{Subst}\left[\frac{x^m F\left[\frac{x^2}{1+x^2}\right]}{\left(1+x^2\right)^{m/2+1}},\,x,\,\text{Tan}[e+fx]\right]\partial_x \text{Tan}[e+fx]$

Rule: If $\frac{m}{2}\in\mathbb{Z}\land\frac{n}{2}\in\mathbb{Z}\land p\in\mathbb{Z}$, then

$$\int \text{Sin}[e+fx]^m \left(a+b\,\text{Sin}[e+fx]^n\right)^p dx \to \frac{1}{f}\,\text{Subst}\left[\int \frac{x^m \left(a\left(1+x^2\right)^{n/2}+b\,x^n\right)^p}{\left(1+x^2\right)^{m/2+n\,p/2+1}}dx,\,x,\,\text{Tan}[e+fx]\right]$$

Program code:

```
Int[sin[e_.+f_.*x_]^m_*(a_+b_.*sin[e_.+f_.*x_]^4)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff^(m+1)/f*Subst[Int[x^m*(a+2*a*ff^2*x^2+(a+b)*ff^4*x^4)^p/(1+ff^2*x^2)^(m/2+2*p+1),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,e,f},x] && IntegerQ[m/2] && IntegerQ[p]
```

```
Int[sin[e_.+f_.*x_]^m_*(a_+b_.*sin[e_.+f_.*x_]^n_)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff^(m+1)/f*Subst[Int[x^m*(a*(1+ff^2*x^2)^(n/2)+b*ff^n*x^n)^p/(1+ff^2*x^2)^(m/2+n*p/2+1),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,e,f},x] && IntegerQ[m/2] && IntegerQ[n/2] && IntegerQ[p]
```

3: $\int \text{Sin}[e+fx]^m \left(a+b\,\text{Sin}[e+fx]^4\right)^p dx$ when $\frac{m}{2}\in\mathbb{Z}\land p-\frac{1}{2}\in\mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $\frac{m}{2}\in\mathbb{Z}$, then $\text{Sin}[z]^m == \frac{\text{Tan}[z]^m}{\left(1+\text{Tan}[z]^2\right)^{m/2}}$

Basis: If $\frac{n}{2} \in \mathbb{Z}$, then $a+b\,\text{Sin}[z]^n == \frac{a\,\text{Sec}[z]^n+b\,\text{Tan}[z]^n}{\text{Sec}[z]^n} == \frac{a\left(1+\text{Tan}[z]^2\right)^{n/2}+b\,\text{Tan}[z]^n}{\left(1+\text{Tan}[z]^2\right)^{n/2}}$

Basis: If $\frac{n}{2} \in \mathbb{Z}$, then $\partial_x \frac{(a+b\,\text{Sin}[e+f\,x]^n)^p\left(\text{Sec}[e+f\,x]^2\right)^{n\,p/2}}{(a\,\text{Sec}[e+f\,x]^n+b\,\text{Tan}[e+f\,x]^n)^p} == 0$

Basis: $F[\text{Tan}[e+f\,x]] == \frac{1}{f}\,\text{Subst}\left[\frac{F[x]}{1+x^2},\,x,\,\text{Tan}[e+f\,x]\right]\partial_x\text{Tan}[e+f\,x]$

Rule: If $\frac{m}{2} \in \mathbb{Z} \land p-\frac{1}{2} \in \mathbb{Z}$, then

$$\int \text{Sin}[e+f\,x]^m\left(a+b\,\text{Sin}[e+f\,x]^4\right)^p dx \rightarrow \frac{\left(a+b\,\text{Sin}[e+f\,x]^4\right)^p\left(\text{Sec}[e+f\,x]^2\right)^{2\,p}}{\left(a\,\text{Sec}[e+f\,x]^4+b\,\text{Tan}[e+f\,x]^4\right)^p}\int\frac{\text{Tan}[e+f\,x]^m\left(a\left(1+\text{Tan}[e+f\,x]^2\right)^2+b\,\text{Tan}[e+f\,x]^4\right)^p}{\left(1+\text{Tan}[e+f\,x]^2\right)^{m/2+2\,p}}dx$$

$$\rightarrow \frac{\left(a+b\,\text{Sin}[e+f\,x]^4\right)^p\left(\text{Sec}[e+f\,x]^2\right)^{2\,p}}{f\left(a\,\text{Sec}[e+f\,x]^4+b\,\text{Tan}[e+f\,x]^4\right)^p}\,\text{Subst}\left[\int\frac{x^m\left(a\left(1+x^2\right)^2+b\,x^4\right)^p}{\left(1+x^2\right)^{m/2+2\,p+1}}dx,\,x,\,\text{Tan}[e+f\,x]\right]$$

–

Program code:

```
Int[sin[e_.+f_.*x_]^m_*(a_+b_.*sin[e_.+f_.*x_]^4)^p_,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff^(m+1)*(a+b*Sin[e+f*x]^4)^p*(Sec[e+f*x]^2)^(2*p)/(f*Apart[a*(1+Tan[e+f*x]^2)^2+b*Tan[e+f*x]^4]^p)*
    Subst[Int[x^m*ExpandToSum[a*(1+ff^2*x^2)^2+b*ff^4*x^4,x]^p/(1+ff^2*x^2)^(m/2+2*p+1),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,e,f,p},x] && IntegerQ[m/2] && IntegerQ[p-1/2]
```

4: $\int \sin[e+f\,x]^m\,(a+b\sin[e+f\,x]^n)^p\,dx$ when $(m\mid p)\in\mathbb{Z}\wedge(n==4\vee p>0\vee p==-1\wedge n\in\mathbb{Z})$

Derivation: Algebraic expansion

Rule: If $(m\mid p)\in\mathbb{Z}\wedge(n==4\vee p>0\vee p==-1\wedge n\in\mathbb{Z})$, then

$$\int \sin[e+f\,x]^m\,(a+b\sin[e+f\,x]^n)^p\,dx\;\rightarrow\;\int \text{ExpandTrig}\left[\sin[e+f\,x]^m\,(a+b\sin[e+f\,x]^n)^p,\,x\right]dx$$

Program code:

```
Int[sin[e_.+f_.*x_]^m_.*(a_+b_.*sin[e_.+f_.*x_]^n_)^p_.,x_Symbol] :=
  Int[ExpandTrig[sin[e+f*x]^m*(a+b*sin[e+f*x]^n)^p,x],x] /;
FreeQ[{a,b,e,f},x] && IntegersQ[m,p] && (EqQ[n,4] || GtQ[p,0] || EqQ[p,-1] && IntegerQ[n])
```

5: $\int (d\sin[e+f\,x])^m\,(a+b\,(c\sin[e+f\,x])^n)^p\,dx$ when $p\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p\in\mathbb{Z}^+$, then

$$\int (d\sin[e+f\,x])^m\,(a+b\,(c\sin[e+f\,x])^n)^p\,dx\;\rightarrow\;\int \text{ExpandTrig}\left[(d\sin[e+f\,x])^m\,(a+b\,(c\sin[e+f\,x])^n)^p,\,x\right]dx$$

Program code:

```
Int[(d_.*sin[e_.+f_.*x_])^m_.*(a_+b_.*(c_.*sin[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  Int[ExpandTrig[(d*sin[e+f*x])^m*(a+b*(c*sin[e+f*x])^n)^p,x],x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && IGtQ[p,0]
```

U: $\int (d \operatorname{Sin}[e+f x])^m \left(a+b (c \operatorname{Sin}[e+f x])^n\right)^p \, dx$

Rule:

$$\int (d \operatorname{Sin}[e+f x])^m \left(a+b (c \operatorname{Sin}[e+f x])^n\right)^p \, dx \longrightarrow \int (d \operatorname{Sin}[e+f x])^m \left(a+b (c \operatorname{Sin}[e+f x])^n\right)^p \, dx$$

Program code:

```
Int[(d_.*sin[e_.+f_.*x_])^m_.*(a_+b_.*(c_.*sin[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  Unintegrable[(d*Sin[e+f*x])^m*(a+b*(c*Sin[e+f*x])^n)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x]
```

4. $\int (d \operatorname{Cos}[e+f x])^m \left(a+b (c \operatorname{Sin}[e+f x])^n\right)^p \, dx$

1: $\int \operatorname{Cos}[e+f x]^m \left(a+b (c \operatorname{Sin}[e+f x])^n\right)^p \, dx$ when $\frac{m-1}{2} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{m-1}{2} \in \mathbb{Z}$, then

$$\operatorname{Cos}[e+f x]^m F[\operatorname{Sin}[e+f x]] = \frac{1}{f} \operatorname{Subst}\left[\left(1-x^2\right)^{\frac{m-1}{2}} F[x], x, \operatorname{Sin}[e+f x]\right] \partial_x \operatorname{Sin}[e+f x]$$

Rule: If $\frac{m-1}{2} \in \mathbb{Z}$, then

$$\int \operatorname{Cos}[e+f x]^m \left(a+b (c \operatorname{Sin}[e+f x])^n\right)^p \, dx \longrightarrow \frac{1}{f} \operatorname{Subst}\left[\int \left(1-x^2\right)^{\frac{m-1}{2}} \left(a+b (c x)^n\right)^p \, dx, x, \operatorname{Sin}[e+f x]\right]$$

Program code:

```
Int[cos[e_.+f_.*x_]^m_.*(a_+b_.*(c_.*sin[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Sin[e+f*x],x]},
  ff/f*Subst[Int[(1-ff^2*x^2)^((m-1)/2)*(a+b*(c*ff*x)^n)^p,x],x,Sin[e+f*x]/ff]] /;
FreeQ[{a,b,c,e,f,n,p},x] && IntegerQ[(m-1)/2] && (EqQ[n,4] || GtQ[m,0] || IGtQ[p,0] || IntegersQ[m,p])
```

2: $\int \text{Cos}[e+f\,x]^m\,(a+b\,\text{Sin}[e+f\,x]^n)^p\,dx$ when $\frac{m}{2}\in\mathbb{Z}\,\land\,\frac{n}{2}\in\mathbb{Z}\,\land\,p\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $\text{Cos}[z]^2 == \frac{1}{1+\text{Tan}[z]^2}$

Basis: $\text{Sin}[z]^2 == \frac{\text{Tan}[z]^2}{1+\text{Tan}[z]^2}$

Basis: If $\frac{m}{2}\in\mathbb{Z}$, then $\text{Cos}[e+f\,x]^m\,F\left[\text{Sin}[e+f\,x]^2\right] == \frac{1}{f}\,\text{Subst}\left[\frac{F\left[\frac{x^2}{1+x^2}\right]}{(1+x^2)^{m/2+1}},\,x,\,\text{Tan}[e+f\,x]\right]\,\partial_x\text{Tan}[e+f\,x]$

Rule: If $\frac{m}{2}\in\mathbb{Z}\,\land\,\frac{n}{2}\in\mathbb{Z}\,\land\,p\in\mathbb{Z}$, then

$$\int \text{Cos}[e+f\,x]^m\,(a+b\,\text{Sin}[e+f\,x]^n)^p\,dx \,\rightarrow\, \frac{1}{f}\,\text{Subst}\left[\int\frac{\left(b\,x^n+a\,(1+x^2)^{n/2}\right)^p}{(1+x^2)^{m/2+n\,p/2+1}}\,dx,\,x,\,\text{Tan}[e+f\,x]\right]$$

Program code:

```
Int[cos[e_.+f_.*x_]^m_*(a_+b_.*sin[e_.+f_.*x_]^4)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff/f*Subst[Int[(a+2*a*ff^2*x^2+(a+b)*ff^4*x^4)^p/(1+ff^2*x^2)^(m/2+2*p+1),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,e,f},x] && IntegerQ[m/2] && IntegerQ[p]
```

```
Int[cos[e_.+f_.*x_]^m_*(a_+b_.*sin[e_.+f_.*x_]^n_)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff/f*Subst[Int[(b*ff^n*x^n+a*(1+ff^2*x^2)^(n/2))^p/(1+ff^2*x^2)^(m/2+n*p/2+1),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,e,f},x] && IntegerQ[m/2] && IntegerQ[n/2] && IntegerQ[p]
```

3. $\int\frac{\text{Cos}[e+f\,x]^m}{a+b\,\text{Sin}[e+f\,x]^n}\,dx$ when $\frac{m}{2}\in\mathbb{Z}\,\land\,\frac{n-1}{2}\in\mathbb{Z}$

1: $\int \frac{\text{Cos}[e+f\,x]^m}{a+b\,\text{Sin}[e+f\,x]^n}\,dx$ when $\frac{m}{2}\in\mathbb{Z}^+\land\frac{n-1}{2}\in\mathbb{Z}$

Derivation: Algebraic expansion

Basis: $\text{Cos}[z]^2 == 1-\text{Sin}[z]^2$

Rule: If $\frac{m}{2}\in\mathbb{Z}^+\land\frac{n-1}{2}\in\mathbb{Z}$, then

$$\int \frac{\text{Cos}[e+f\,x]^m}{a+b\,\text{Sin}[e+f\,x]^n}\,dx \longrightarrow \int \text{Expand}\left[\frac{\left(1-\text{Sin}[e+f\,x]^2\right)^{m/2}}{a+b\,\text{Sin}[e+f\,x]^n},\,x\right]dx$$

Program code:

```
Int[cos[e_.+f_.*x_]^m_/(a_+b_.*sin[e_.+f_.*x_]^n_),x_Symbol] :=
  Int[Expand[(1-Sin[e+f*x]^2)^(m/2)/(a+b*Sin[e+f*x]^n),x],x] /;
FreeQ[{a,b,e,f},x] && IGtQ[m/2,0] && IntegerQ[(n-1)/2]
```

x: $\int \frac{\text{Cos}[e+f x]^m}{a+b\,\text{Sin}[e+f x]^n}\,dx$ when $\frac{m}{2}\in\mathbb{Z}\land\frac{n-1}{2}\in\mathbb{Z}\land p-1\in\mathbb{Z}^-\land m<0$

Derivation: Algebraic expansion

Basis: $\text{Cos}[z]^2 = 1-\text{Sin}[z]^2$

Rule: If $\frac{m}{2}\in\mathbb{Z}\land\frac{n-1}{2}\in\mathbb{Z}\land p-1\in\mathbb{Z}^-\land m<0$, then

$$\int \frac{\text{Cos}[e+f x]^m}{a+b\,\text{Sin}[e+f x]^n}\,dx \longrightarrow \int \text{ExpandTrig}\left[\frac{\left(1-\text{Sin}[e+f x]^2\right)^{m/2}}{a+b\,\text{Sin}[e+f x]^n},\,x\right]dx$$

Program code:

```
(* Int[cos[e_.+f_.*x_]^m_*(a_+b_.*sin[e_.+f_.*x_]^n_)^p_,x_Symbol] :=
  Int[ExpandTrig[(1-sin[e+f*x]^2)^(m/2)*(a+b*sin[e+f*x]^n)^p,x],x] /;
FreeQ[{a,b,e,f},x] && IntegerQ[m/2] && IntegerQ[(n-1)/2] && ILtQ[p,-1] && LtQ[m,0] *)
```

4: $\int (d\,\text{Cos}[e+f x])^m \left(a+b\,(c\,\text{Sin}[e+f x])^n\right)^p dx$ when $p\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p\in\mathbb{Z}^+$, then

$$\int (d\,\text{Cos}[e+f x])^m \left(a+b\,(c\,\text{Cos}[e+f x])^n\right)^p dx \longrightarrow \int \text{ExpandTrig}\left[(d\,\text{Cos}[e+f x])^m \left(a+b\,(c\,\text{Sin}[e+f x])^n\right)^p,\,x\right]dx$$

Program code:

```
Int[(d_.*cos[e_.+f_.*x_])^m_.*(a_+b_.*(c_.*sin[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  Int[ExpandTrig[(d*cos[e+f*x])^m*(a+b*(c*sin[e+f*x])^n)^p,x],x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && IGtQ[p,0]
```

U: $\int (d \operatorname{Cos}[e + f x])^m (a + b (c \operatorname{Sin}[e + f x])^n)^p \, dx$

Rule:

$$\int (d \operatorname{Cos}[e + f x])^m (a + b (c \operatorname{Sin}[e + f x])^n)^p \, dx \longrightarrow \int (d \operatorname{Cos}[e + f x])^m (a + b (c \operatorname{Sin}[e + f x])^n)^p \, dx$$

Program code:

```
Int[(d_.*cos[e_.+f_.*x_])^m_.*(a_+b_.*(c_.*sin[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  Unintegrable[(d*Cos[e+f*x])^m*(a+b*(c*Sin[e+f*x])^n)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x]
```

5. $\int (d\,\mathrm{Tan}[e+f\,x])^m\,(a+b\,(c\,\mathrm{Sin}[e+f\,x])^n)^p\,dx$

1. $\int \mathrm{Tan}[e+f\,x]^m\,(a+b\,(c\,\mathrm{Sin}[e+f\,x])^n)^p\,dx$ when $\frac{m-1}{2}\in\mathbb{Z}$

1: $\int \mathrm{Tan}[e+f\,x]^m\,(a+b\,\mathrm{Sin}[e+f\,x]^n)^p\,dx$ when $\frac{m-1}{2}\in\mathbb{Z}\wedge\frac{n}{2}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $\mathrm{Tan}[z]^2 = \frac{\mathrm{Sin}[z]^2}{1-\mathrm{Sin}[z]^2}$

Basis: If $\frac{m-1}{2}\in\mathbb{Z}$, then

$$\mathrm{Tan}[e+f\,x]^m\,F\left[\mathrm{Sin}[e+f\,x]^2\right] = \frac{1}{2f}\,\mathrm{Subst}\left[\frac{x^{\frac{m-1}{2}}\,F[x]}{(1-x)^{\frac{m+1}{2}}},\,x,\,\mathrm{Sin}[e+f\,x]^2\right]\,\partial_x\mathrm{Sin}[e+f\,x]^2$$

Rule: If $\frac{m-1}{2}\in\mathbb{Z}\wedge\frac{n}{2}\in\mathbb{Z}$, then

$$\int \mathrm{Tan}[e+f\,x]^m\,(a+b\,\mathrm{Sin}[e+f\,x]^n)^p\,dx \to \frac{1}{2f}\,\mathrm{Subst}\left[\int\frac{x^{\frac{m-1}{2}}\,(a+b\,x^{n/2})^p}{(1-x)^{\frac{m+1}{2}}}\,dx,\,x,\,\mathrm{Sin}[e+f\,x]^2\right]$$

Program code:

```
Int[tan[e_.+f_.*x_]^m_.*(a_+b_.*sin[e_.+f_.*x_]^n_)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Sin[e+f*x]^2,x]},
  ff^((m+1)/2)/(2*f)*Subst[Int[x^((m-1)/2)*(a+b*ff^(n/2)*x^(n/2))^p/(1-ff*x)^((m+1)/2),x],x,Sin[e+f*x]^2/ff]] /;
FreeQ[{a,b,e,f,p},x] && IntegerQ[(m-1)/2] && IntegerQ[n/2]
```

2: $\int \mathrm{Tan}[e+f\,x]^m\,(a+b\,(c\,\mathrm{Sin}[e+f\,x])^n)^p\,dx$ when $\frac{m-1}{2}\in\mathbb{Z}^-$

Derivation: Integration by substitution

Basis: $\mathrm{Tan}[z]^2 = \frac{\mathrm{Sin}[z]^2}{1-\mathrm{Sin}[z]^2}$

Basis: If $\frac{m-1}{2}\in\mathbb{Z}$, then $\mathrm{Tan}[e+f\,x]^m\,F[\mathrm{Sin}[e+f\,x]] = \frac{1}{f}\,\mathrm{Subst}\left[\frac{x^m\,F[x]}{(1-x^2)^{\frac{m+1}{2}}},\,x,\,\mathrm{Sin}[e+f\,x]\right]\,\partial_x\mathrm{Sin}[e+f\,x]$

Rule: If $\frac{m-1}{2} \in \mathbb{Z}^-$, then

$$\int \mathrm{Tan}[e+f\,x]^m\,\left(a+b\,\left(c\,\mathrm{Sin}[e+f\,x]\right)^n\right)^p\,dx \longrightarrow \frac{1}{f}\,\mathrm{Subst}\left[\int \frac{x^m\,\left(a+b\,(c\,x)^n\right)^p}{\left(1-x^2\right)^{\frac{m+1}{2}}}\,dx,\,x,\,\mathrm{Sin}[e+f\,x]\right]$$

Program code:

```
Int[tan[e_.+f_.*x_]^m_.*(a_+b_.*(c_.*sin[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Sin[e+f*x],x]},
  ff^(m+1)/f*Subst[Int[x^m*(a+b*(c*ff*x)^n)^p/(1-ff^2*x^2)^((m+1)/2),x],x,Sin[e+f*x]/ff]] /;
FreeQ[{a,b,c,e,f,n,p},x] && ILtQ[(m-1)/2,0]
```

2. $\int (d\,\text{Tan}[e+f\,x])^m\,(a+b\,\text{Sin}[e+f\,x]^n)^p\,dx$ when $\frac{n}{2}\in\mathbb{Z}$

1: $\int (d\,\text{Tan}[e+f\,x])^m\,(a+b\,\text{Sin}[e+f\,x]^4)^p\,dx$ when $p\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $\text{Sin}[z]^2 = \frac{\text{Tan}[z]^2}{1+\text{Tan}[z]^2}$

Basis: $(d\,\text{Tan}[e+f\,x])^m\,F\left[\text{Sin}[e+f\,x]^2\right] = \frac{1}{f}\,\text{Subst}\left[\frac{(d\,x)^m\,F\left[\frac{x^2}{1+x^2}\right]}{1+x^2},\,x,\,\text{Tan}[e+f\,x]\right]\,\partial_x\,\text{Tan}[e+f\,x]$

Rule: If $p\in\mathbb{Z}$, then

$$\int (d\,\text{Tan}[e+f\,x])^m\,(a+b\,\text{Sin}[e+f\,x]^4)^p\,dx \to \frac{1}{f}\,\text{Subst}\left[\int \frac{(d\,x)^m\,\left(a\,(1+x^2)^2+b\,x^4\right)^p}{(1+x^2)^{2\,p+1}}\,dx,\,x,\,\text{Tan}[e+f\,x]\right]$$

Program code:

```
Int[(d_.*tan[e_.+f_.*x_])^m_*(a_+b_.*sin[e_.+f_.*x_]^4)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff/f*Subst[Int[(d*ff*x)^m*ExpandToSum[a*(1+ff^2*x^2)^2+b*ff^4*x^4,x]^p/(1+ff^2*x^2)^(2*p+1),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,d,e,f,m},x] && IntegerQ[p]
```

2: $\int (d \operatorname{Tan}[e+f x])^m \left(a+b \operatorname{Sin}[e+f x]^4\right)^p dx$ when $p-\frac{1}{2} \in \mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $\frac{n}{2} \in \mathbb{Z}$, then $a+b \operatorname{Sin}[z]^n == \frac{a \operatorname{Sec}[z]^n+b \operatorname{Tan}[z]^n}{\operatorname{Sec}[z]^n} == \frac{a\left(1+\operatorname{Tan}[z]^2\right)^{n/2}+b \operatorname{Tan}[z]^n}{\left(1+\operatorname{Tan}[z]^2\right)^{n/2}}$

Basis: If $\frac{n}{2} \in \mathbb{Z}$, then $\partial_x \frac{(a+b \operatorname{Sin}[e+f x]^n)^p \left(\operatorname{Sec}[e+f x]^2\right)^{n p/2}}{(a \operatorname{Sec}[e+f x]^n+b \operatorname{Tan}[e+f x]^n)^p} == 0$

Basis: $F[\operatorname{Tan}[e+f x]] == \frac{1}{f} \operatorname{Subst}\left[\frac{F[x]}{1+x^2}, x, \operatorname{Tan}[e+f x]\right] \partial_x \operatorname{Tan}[e+f x]$

Rule: If $p-\frac{1}{2} \in \mathbb{Z}$, then

$$\int (d \operatorname{Tan}[e+f x])^m \left(a+b \operatorname{Sin}[e+f x]^4\right)^p dx \rightarrow \frac{\left(a+b \operatorname{Sin}[e+f x]^4\right)^p \left(\operatorname{Sec}[e+f x]^2\right)^{2p}}{\left(a \operatorname{Sec}[e+f x]^4+b \operatorname{Tan}[e+f x]^4\right)^p} \int \frac{(d \operatorname{Tan}[e+f x])^m \left(a\left(1+\operatorname{Tan}[e+f x]^2\right)^2+b \operatorname{Tan}[e+f x]^5\right)^p}{\left(1+\operatorname{Tan}[e+f x]^2\right)^{2p+1}} dx$$

$$\rightarrow \frac{\left(a+b \operatorname{Sin}[e+f x]^4\right)^p \left(\operatorname{Sec}[e+f x]^2\right)^{2p}}{f\left(a \operatorname{Sec}[e+f x]^4+b \operatorname{Tan}[e+f x]^4\right)^p} \operatorname{Subst}\left[\int \frac{(d x)^m \left(a\left(1+x^2\right)^2+b x^4\right)^p}{\left(1+x^2\right)^{2p+1}} dx, x, \operatorname{Tan}[e+f x]\right]$$

Program code:

```
Int[(d_.*tan[e_.+f_.*x_])^m_*(a_+b_.*sin[e_.+f_.*x_]^4)^p_,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff*(a+b*Sin[e+f*x]^4)^p*(Sec[e+f*x]^2)^(2*p)/(f*Apart[a*(1+Tan[e+f*x]^2)^2+b*Tan[e+f*x]^4]^p)*
    Subst[Int[(d*ff*x)^m*ExpandToSum[a*(1+ff^2*x^2)^2+b*ff^4*x^4,x]^p/(1+ff^2*x^2)^(2*p+1),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,d,e,f,m},x] && IntegerQ[p-1/2]
```

3: $\int (d \operatorname{Tan}[e+f x])^m (a+b \operatorname{Sin}[e+f x]^n)^p \, dx$ when $\frac{n}{2} \in \mathbb{Z} \land p \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $\operatorname{Sin}[z]^2 == \frac{\operatorname{Tan}[z]^2}{1+\operatorname{Tan}[z]^2}$

Basis: $(d \operatorname{Tan}[e+f x])^m F\left[\operatorname{Sin}[e+f x]^2\right] == \frac{1}{f} \operatorname{Subst}\left[\frac{(d x)^m F\left[\frac{x^2}{1+x^2}\right]}{1+x^2}, x, \operatorname{Tan}[e+f x]\right] \partial_x \operatorname{Tan}[e+f x]$

Rule: If $\frac{n}{2} \in \mathbb{Z} \land p \in \mathbb{Z}^+$, then

$$\int (d \operatorname{Tan}[e+f x])^m (a+b \operatorname{Sin}[e+f x]^n)^p \, dx \to \frac{1}{f} \operatorname{Subst}\left[\int \frac{(d x)^m \left(b x^n + a \left(1+x^2\right)^{n/2}\right)^p}{\left(1+x^2\right)^{n p/2+1}} \, dx, x, \operatorname{Tan}[e+f x]\right]$$

Program code:

```
Int[(d_.*tan[e_.+f_.*x_])^m_*(a_+b_.*sin[e_.+f_.*x_]^n_)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff^(m+1)/f*Subst[Int[(d*x)^m*(b*ff^n*x^n+a*(1+ff^2*x^2)^(n/2))^p/(1+ff^2*x^2)^(n*p/2+1),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,d,e,f,m},x] && IntegerQ[n/2] && IGtQ[p,0]
```

3: $\int (d \operatorname{Tan}[e+f x])^m \left(a+b (c \operatorname{Sin}[e+f x])^n\right)^p dx$ when $p \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}^+$, then

$$\int (d \operatorname{Tan}[e+f x])^m \left(a+b (c \operatorname{Sin}[e+f x])^n\right)^p dx \longrightarrow \int \operatorname{ExpandTrig}\left[(d \operatorname{Tan}[e+f x])^m \left(a+b (c \operatorname{Sin}[e+f x])^n\right)^p, x\right] dx$$

Program code:

```
Int[(d_.*tan[e_.+f_.*x_])^m_.*(a_+b_.*(c_.*sin[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  Int[ExpandTrig[(d*tan[e+f*x])^m*(a+b*(c*sin[e+f*x])^n)^p,x],x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && IGtQ[p,0]
```

U: $\int (d \operatorname{Tan}[e+f x])^m \left(a+b (c \operatorname{Sin}[e+f x])^n\right)^p dx$

Rule:

$$\int (d \operatorname{Tan}[e+f x])^m \left(a+b (c \operatorname{Sin}[e+f x])^n\right)^p dx \longrightarrow \int (d \operatorname{Tan}[e+f x])^m \left(a+b (c \operatorname{Sin}[e+f x])^n\right)^p dx$$

Program code:

```
Int[(d_.*tan[e_.+f_.*x_])^m_.*(a_+b_.*(c_.*sin[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  Unintegrable[(d*Tan[e+f*x])^m*(a+b*(c*Sin[e+f*x])^n)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x]
```

6: $\int (d\,\text{Cot}[e+f\,x])^m\,(a+b\,(c\,\text{Sin}[e+f\,x])^n)^p\,dx$ when $m\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\left((d\,\text{Cot}[e+f\,x])^m\left(\frac{\text{Tan}[e+f\,x]}{d}\right)^m\right)=0$

Rule: If $m\notin\mathbb{Z}$, then

$$\int (d\,\text{Cot}[e+f\,x])^m\,(a+b\,(c\,\text{Sin}[e+f\,x])^n)^p\,dx\;\to\;(d\,\text{Cot}[e+f\,x])^{\text{FracPart}[m]}\left(\frac{\text{Tan}[e+f\,x]}{d}\right)^{\text{FracPart}[m]}\int\left(\frac{\text{Tan}[e+f\,x]}{d}\right)^{-m}(a+b\,(c\,\text{Sin}[e+f\,x])^n)^p\,dx$$

Program code:

```
Int[(d_.*cot[e_.+f_.*x_])^m_*(a_+b_.*(c_.*sin[e_.+f_.*x_])^n_)^p_,x_Symbol] :=
  (d*Cot[e+f*x])^FracPart[m]*(Tan[e+f*x]/d)^FracPart[m]*Int[(Tan[e+f*x]/d)^(-m)*(a+b*(c*Sin[e+f*x])^n)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && Not[IntegerQ[m]]
```

7: $\int (d\,\text{Sec}[e+f\,x])^m\,(a+b\,(c\,\text{Sin}[e+f\,x])^n)^p\,dx$ when $m\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\left((d\,\text{Sec}[e+f\,x])^m\left(\frac{\text{Cos}[e+f\,x]}{d}\right)^m\right)=0$

Rule: If $m\notin\mathbb{Z}$, then

$$\int (d\,\text{Sec}[e+f\,x])^m\,(a+b\,(c\,\text{Sin}[e+f\,x])^n)^p\,dx\;\to\;(d\,\text{Sec}[e+f\,x])^{\text{FracPart}[m]}\left(\frac{\text{Cos}[e+f\,x]}{d}\right)^{\text{FracPart}[m]}\int\left(\frac{\text{Cos}[e+f\,x]}{d}\right)^{-m}(a+b\,(c\,\text{Sin}[e+f\,x])^n)^p\,dx$$

Program code:

```
Int[(d_.*sec[e_.+f_.*x_])^m_*(a_+b_.*(c_.*sin[e_.+f_.*x_])^n_)^p_,x_Symbol] :=
  (d*Sec[e+f*x])^FracPart[m]*(Cos[e+f*x]/d)^FracPart[m]*Int[(Cos[e+f*x]/d)^(-m)*(a+b*(c*Sin[e+f*x])^n)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && Not[IntegerQ[m]]
```

8. $\int (d \operatorname{Csc}[e + f x])^m (a + b (c \operatorname{Sin}[e + f x])^n)^p \, dx$ when $m \notin \mathbb{Z}$

1: $\int (d \operatorname{Csc}[e + f x])^m (a + b \operatorname{Sin}[e + f x]^n)^p \, dx$ when $m \notin \mathbb{Z} \land (n \mid p) \in \mathbb{Z}$

Derivation: Algebraic normalization

Basis: If $(n \mid p) \in \mathbb{Z}$, then $(a + b \operatorname{Sin}[e + f x]^n)^p == d^{n p} (d \operatorname{Csc}[e + f x])^{-n p} (b + a \operatorname{Csc}[e + f x]^n)^p$

Rule: If $m \notin \mathbb{Z} \land (n \mid p) \in \mathbb{Z}$, then

$$\int (d \operatorname{Csc}[e + f x])^m (a + b \operatorname{Sin}[e + f x]^n)^p \, dx \longrightarrow d^{n p} \int (d \operatorname{Csc}[e + f x])^{m-n p} (b + a \operatorname{Csc}[e + f x]^n)^p \, dx$$

Program code:

```
Int[(d_.*csc[e_.+f_.*x_])^m_*(a_+b_.*sin[e_.+f_.*x_]^n_.)^p_.,x_Symbol] :=
  d^(n*p)*Int[(d*Csc[e+f*x])^(m-n*p)*(b+a*Csc[e+f*x]^n)^p,x] /;
FreeQ[{a,b,d,e,f,m,n,p},x] && Not[IntegerQ[m]] && IntegersQ[n,p]
```

2: $\int (d\,\text{Csc}[e+f\,x])^m\,(a+b\,(c\,\text{Sin}[e+f\,x])^n)^p\,dx$ when $m\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\left((d\,\text{Csc}[e+f\,x])^m\left(\frac{\text{Sin}[e+f\,x]}{d}\right)^m\right)=0$

–

Rule: If $m\notin\mathbb{Z}$, then

$$\int (d\,\text{Csc}[e+f\,x])^m\,(a+b\,(c\,\text{Sin}[e+f\,x])^n)^p\,dx \longrightarrow (d\,\text{Csc}[e+f\,x])^{\text{FracPart}[m]}\left(\frac{\text{Sin}[e+f\,x]}{d}\right)^{\text{FracPart}[m]}\int\left(\frac{\text{Sin}[e+f\,x]}{d}\right)^{-m}(a+b\,(c\,\text{Sin}[e+f\,x])^n)^p\,dx$$

–

Program code:

```
Int[(d_.*csc[e_.+f_.*x_])^m_*(a_+b_.*(c_.*sin[e_.+f_.*x_])^n_)^p_,x_Symbol] :=
  (d*Csc[e+f*x])^FracPart[m]*(Sin[e+f*x]/d)^FracPart[m]*Int[(Sin[e+f*x]/d)^(-m)*(a+b*(c*Sin[e+f*x])^n)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && Not[IntegerQ[m]]
```

Rules for integrands of the form $\left(a+b\,(c\,\mathrm{Sin}[e+f\,x]+d\,\mathrm{Cos}[e+f\,x])^2\right)^p$

1. $\int\left(a+b\,(c\,\mathrm{Sin}[e+f\,x]+d\,\mathrm{Cos}[e+f\,x])^2\right)^p\,dx$ when $p^2==\frac{1}{4}$

1: $\int\left(a+b\,(c\,\mathrm{Sin}[e+f\,x]+d\,\mathrm{Cos}[e+f\,x])^2\right)^p\,dx$ when $p^2==\frac{1}{4}\land a>0$

Derivation: Algebraic simplification

Basis: $c\,\mathrm{Sin}[z]+d\,\mathrm{Cos}[z]==\sqrt{c^2+d^2}\,\mathrm{Sin}[\mathrm{ArcTan}[c,d]+z]$

Rule: If $p^2==\frac{1}{4}\land a>0$, then

$$\int\left(a+b\,(c\,\mathrm{Sin}[e+f\,x]+d\,\mathrm{Cos}[e+f\,x])^2\right)^p\,dx\longrightarrow\int\left(a+b\left(\sqrt{c^2+d^2}\,\mathrm{Sin}[\mathrm{ArcTan}[c,d]+e+f\,x]\right)^2\right)^p\,dx$$

Program code:

```
Int[(a_+b_.*(c_.*sin[e_.+f_.*x_]+d_.*cos[e_.+f_.*x_])^2)^p_,x_Symbol] :=
  Int[(a+b*(Sqrt[c^2+d^2]*Sin[ArcTan[c,d]+e+f*x])^2)^p,x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[p^2,1/4] && GtQ[a,0]
```

2: $\int \left(a+b\,(c\,\mathrm{Sin}[e+f\,x]+d\,\mathrm{Cos}[e+f\,x])^2\right)^p\,dx$ when $p^2 == \frac{1}{4} \land a \not> 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{a+b\,(c\,\mathrm{Sin}[e+f\,x]+d\,\mathrm{Cos}[e+f\,x])^2}}{\sqrt{1+\frac{b\,(c\,\mathrm{Sin}[e+f\,x]+d\,\mathrm{Cos}[e+f\,x])^2}{a}}} == 0$

Rule: If $p^2 == \frac{1}{4} \land a \not> 0$, then

$$\int \left(a+b\,(c\,\mathrm{Sin}[e+f\,x]+d\,\mathrm{Cos}[e+f\,x])^2\right)^p\,dx \to \frac{\left(a+b\,(c\,\mathrm{Sin}[e+f\,x]+d\,\mathrm{Cos}[e+f\,x])^2\right)^p}{\left(1+\frac{b\,(c\,\mathrm{Sin}[e+f\,x]+d\,\mathrm{Cos}[e+f\,x])^2}{a}\right)^p}\int \left(1+\frac{b\,(c\,\mathrm{Sin}[e+f\,x]+d\,\mathrm{Cos}[e+f\,x])^2}{a}\right)^p\,dx$$

–

Program code:

```
Int[(a_+b_.*(c_.*sin[e_.+f_.*x_]+d_.*cos[e_.+f_.*x_])^2)^p_,x_Symbol] :=
  (a+b*(c*Sin[e+f*x]+d*Cos[e+f*x])^2)^p/(1+(b*(c*Sin[e+f*x]+d*Cos[e+f*x])^2)/a)^p*
    Int[(1+(b*(c*Sin[e+f*x]+d*Cos[e+f*x])^2)/a)^p,x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[p^2,1/4] && Not[GtQ[a,0]]
```